Trečiojo etapo antroji dalis

Jaunesniųjų grupė

160. Šokolado valgymas. Įsivaizduokime, kad šokolado gabalas nėra kvadrato formos. Tuomet galime atlaužti tokį šokolado gabalą, kad likęs gabalas būtų kvadrato formos. 

160.1 pav. 

Kaip belaužtų šokoladą priešininkas, gabalas vėl taps nebe kvadratinis. 

160.2 pav.

Taip priešininkui kiekvieną kartą atiteks kvadrato formos šokolado gabalas, o mums – stačiakampis (ne kvadratinis). 

Kadangi plytelė „apgavikė“ yra kvadrato formos (t. y. vienas kvadratėlis), ji ir atiteks priešininkui.

Na o jei turimas gabalas yra kvadrato formos, laimėti neteks. Tuomet laužiame kuo daugiau šokolado: paliekame tik vieną plytelių eilutę ar stulpelį, stengiantis, kad liktų kuo mažiau plytelių.

program šokolado_valgymas;

  var n, m: integer;

begin
  readln (n, m);

  if n < m { jei duotasis šokoladas ne kvadrato formos }
     then writeln ('L V ', m-n) { laimi pirmasis }
     else if n > m then writeln ('L H ',n-m )

                   else writeln ('P H ', n-1); { laimi antrasis }
end.

	Testo nr.
	Pradiniai duomenys
	Rezultatai
	Paaiškinimai

	1
	1 5
	L V 4
	Paprastas atvejis

	2
	7 7
	P V 6
	Šokoladas kvadrato formos ;

Kitas galimas atsakymas: P H 6.

	3
	1456 1456
	P V 1455
	Didelis kvadrato formos šokoladas;

Kitas galimas atsakymas: P H 1455

	4
	3 4
	L V 1
	Sąlygoje pateiktas pavyzdys

	5
	9 10
	L V 1
	Ribinis atvejis (m = n + 1)

	6
	1789 1788
	L H 1
	Ribinis atvejis (n = m + 1)

	7
	45 2025
	L V 1980
	7-9 testai: keletas atvejų, 

	8
	2420 31
	L H 2389
	kai laimi pirmasis

	9
	82 2296
	L V 2214
	

	10
	32767 458
	L H 32309
	Ribinis atvejis – (n = maxint)


161. Buratinas, lapė Alisa ir katinas Bazilijus kvailių lauke. Iš pirmo žvilgsnio uždavinys atrodo gan paprastas: tereikia išnagrinėti visus įmanomus variantus ir atitinkamai perstumti konteinerius. Juk neprašoma rasti trumpiausio kelio. Tačiau pradėjus nagrinėti galimus variantus paaiškėja, kad jų yra labai daug: plėšikai gali būti pasislėpę ties ta pačia arba ties skirtingomis horizontalėmis, jie gali būti vienoje arba skirtingose lauko pusėse, konteineriai gali būti sustatyti greta arba ne greta, jie gali užstoti vieną ar du plėšikus, gali neužstoti ir t.t. Pradėjus nagrinėti visus variantus, uždavinys įgyja dirbtinio intelekto bruožų: veiksmus, kuriuos nesunku atlikti bet kuriam vaikui (galima įsivaizduoti, kad tai yra stalo žaidimas), yra labai sunku suformuluoti algoritmo pavidalu. 

Šiam uždaviniui ieškojome bendresnio sprendimo. Vienas galimų variantų – paieškos į plotį ir Bangos algoritmo taikymas. Šie abu metodai išaiškinti I šio uždavinyno dalies 83 uždavinyje.

Algoritmas būtų toks: bangos metodu nestumdydamas konteinerių Buratinas iš pradinės padėties bando pasiekti raktelį. 

Jei nepasiseka, jis bando eiti visomis keturiomis kryptimis: į kairę, į dešinę, į viršų ir į apačią, suprantama, jei galima paeiti ta kryptimi (t.y. neišeinama už lauko ribų, nepakliūnama į puolamą langelį, neužlipama ant konteinerio). Paėjus kažkuria kryptimi, bangos metodu patikrinama ar iš naujos padėties galima pasiekti raktelį. Jei taip – kelias rastas, jei ne – nauja padėtis įsimenama eilėje ir bandoma eiti kita kryptimi. Išbandžius visas keturias kryptis ir visas naujai gautas padėtis įsiminus eilėje, imama eilės pradžioje esanti padėtis ir iš jos vėl bandoma eiti visomis kryptimis. Veiksmai tęsiami kol randamas sprendinys.

Kadangi nei raktelis, nei plėšikai nejuda, tai įsimenant padėtį eilėje užtenka įsiminti tik Buratino ir abiejų konteinerių koordinates. Taip pat būtina atsiminti iš kurios eilėje esančios padėties gavome šią naują padėtį.

Šį algoritmą pateiksime pseudo–Paskalio kalba.

RAKTELIO_PAIEŠKA (n { lentos dydis }

                  b, k1, k2, p1, p2,  { Buratino, konteinerių, plėšikų }

                  r)                 { bei raktelio koordinatės }
  var rasta; { ar pasiektas raktelis }
      E;     { eilė }
begin

  galva :( 1; uodega :( 1; { eilė kol kas tuščia }
  Įtraukti_į_eilę (E, 0, b, k1, k2, uodega); {įtraukiame pradinę padėtį } 
  { gal raktelis pasiekiamas nestumdant konteinerių }
  BANGA (n, b, k1, k2, p1, p2, rasta);

  if not rasta then { jei raktelis dar nerastas }
   while not rasta do 

    begin
      Paimti_iš_eilės (b, k1, k2, galva);

      while nepaeita keturiomis kryptimis and not rasta do
        if Buratinas gali paeiti atitinkama kryptimi then
         begin 

           Buratinas paeina atitinkama kryptimi, jei reikia pastumdamas konteinerį.
           { gal iš naujos padėties įmanoma pasiekti raktelį } 

           BANGA(n, b_nauja, k1_nauja, k2_nauja, p1, p2, rasta);

           if tokios padėties dar nebuvo eilėje then 
              Įtraukti_į_eilę (galva, b_nauja, 

                               k1_nauja, k2_nauja, uodega);

         end;

        galva :( galva + 1; 

    end;

  { pagal eilėje esančias padėtis naudodamiesi Bangos algoritmu randame Buratino ėjimus }
  gauti_rezultatus(E, uodega, ėjimai, ėjimų_skaičius);

return (ėjimai, ėjimų_skaičius);

Pabandykime įvertinti kokio ilgumo gali būti eilė. Lenta daugiausia gali turėti 16(16 ( 256 langelius. Ir Buratinas ir abu konteineriai gali būti beveik kiekviename šių langelių. Taigi, apytiksliai būtų 256(256(256(16777216 padėtys. 

Tai pakankamai didelis skaičius. Pamėginsime jį sumažinti. Pirmiausia pastebėsime, kad vilčių surasti raktelį gali atsirasti tik pastūmus konteinerį. Jei dabartinėje situacijoje Buratinas negali pasiekti raktelio, tai jam paėjus, bet nepastūmus konteinerio, jis tikrai negalės pasiekti raktelio. Be to, jei Buratinas atliko keletą ėjimų lentoje, bet nepasikeitė konteinerių vieta, Bangos metodu lengvai galime rasti Buratino kelią iš vieno langelio į kitą.

Taigi, vietoj to, kad Buratinas bandytų eiti visomis įmanomomis kryptimis, šiomis kryptimis bandysime pastumti kiekvieną konteinerį. Langelį, kuriame yra Buratinas taip pat reikės įsiminti. Tai iliustruojama žemiau pateiktuose paveikslėliuose (161.3 pav.).

161.3.a pav.

Pirmojo konteinerio Buratinas pastumti negali. Antrąjį konteinerį Buratinas gali pastumti trimis kryptimis – į viršų, į kairę ir į dešinę (161.3.b–d. pav.). Žemyn pastumti antrojo konteinerio negali todėl, kad iš esamos (161.3.a. pav.) padėties Buratinas negali pasiekti langelio (3, 3) iš kurio galėtų antrąjį konteinerį stumti žemyn.

161.3.b pav.

161.3.c pav.

161.3.d pav.

	Eilė
	0, b(1, 1); k1 (2, 1); k2 (2, 3);
	1, b(2, 3); k1 (2, 1); k2 (2, 2);
	1, b(2, 3); k1 (2, 1); k2 (2, 4);
	1, b(2, 3); k1 (2, 1); k2 (3, 3);

	
	1
	2
	3
	4


Iš paskutiniosios (ketvirtosios) padėties Buratinas gali pasiekti raktelį nebestumdydamas konteinerių. Raktelis surastas. Jei tai nepavyktų, būtų imamas antrasis eilės elementas (t. y. padėtis, pavaizduota 161.3.b. pav.) ir būtų bandoma abu konteinerius stumti į visas puses.  

Įtraukiant naują padėtį į eilę, ji įtraukiama, jei anksčiau eilėje nebuvo identiškos padėties. Dvi padėtys laikomos identiškomis, jei konteineriai yra tuose pačiuose laukeliuose, o Buratinas iš vienos padėties langelio nestumdydamas konteinerių gali pasiekti antrosios padėties Buratino langelį. 161.4.a–b. pav.

161.4.a. pav. Ši padėtis identiškai tai padėčiai, kuri pavaizduota 161.3.a pav. Mat Buratinas iš langelio (1, 1) nestumdydamas konteinerių gali pasiekti langelį (2, 4). 

161.4.a. pav. Ši padėtis nėra identiška 161.3.a pav. pavaizduotai padėčiai, nes Buratinas iš langelio (1, 1) nestumdydamas konteinerių negali pasiekti langelio (4, 3). 

Patobulinus algoritmą galimų variantų skaičius sumažėja iki (256(256(65536. Tačiau ir tai yra pakankamai didelis skaičius. Algoritmas bus ypač lėtas, jei Buratinas su abiem konteineriais bus lauko apačioje, o abu plėšikai – viršuje. Įvedame dar vieną patobulinimą: konteineris yra pastumiamas ne per vieną langelį, o tol kol galima. T. y. kol konteineris atsiremia į lauko kraštą arba kitą konteinerį, arba pasiekia horizontalę, kurioje yra plėšikas ir dėl to Buratinas nebegali jo stumti toliau. Taip pakeitus, galimų variantų skaičius bus labai mažas ir gausime efektyvų algoritmą.

Kadangi Paskaliu užrašytas algoritmas yra labai ilgas, jį pateiksime dalimis }.

Pradėsime nuo svarbiausių procedūrų

program buratinas;

  const MAX_E = 1000;   { maksimalus ėjimų skaičius }
        MAX_EIL = 4000; { maksimalus eilės dydis }
        MAX_N = 16;     { maksimalus lentos dydis }
        DX:  array [1..4] of integer = (1, 0, -1, 0);

        DY:  array [1..4] of integer = (0, -1, 0, 1);

  type langelis = record
                    eil, stu: integer;

                  end;

       ėjimai = array [1..MAX_E] of char;

       lenta = array [1..MAX_N, 1..MAX_N] of integer;

       eilė = array [1..MAX_EIL] of record
                                      b, k1, k2: langelis;

                                      iš: integer;

                                    end;

  { žemiau aprašyti globalieji kintamieji }
  var n: integer;       { lentos dydis }
      r,                { raktelio koordinatės }
      p1, p2: langelis; { abiejų plėšikų koordinatės }
      ėj: ėjimai;       { Buratino ėjimai }
      sk: integer;      { jų skaičius }
      Q: eilė;          { eilė }
  {–––––––––––––––––––––––––––––––––pagalbinės funkcijos––––––––––––––––––––}
  function lygu (lan1, lan2: langelis): boolean;

    { patikrina, ar du langeliai sutampa }
  begin
    lygu := (lan1.eil = lan2.eil) and (lan1.stu = lan2.stu);

  end; { lygu }
  function lentoje (lan: langelis): boolean;

  begin
    lentoje := (lan.eil >= 1) and (lan.eil <= n) and
               (lan.stu >= 1) and (lan.stu <= n);

  end; { lentoje }
  function laisva (lan, k1, k2: langelis): boolean;

  begin
    laisva := not lygu(lan, k1) and not lygu (lan, k2);

  end; { laisva }
  function puolamas (lan, k1, k2: langelis): boolean;

    { patikrina, ar langelis yra puolamas }
    var puola1, puola2, gina_k1_p1, gina_k1_p2, gina_k2_p1,

        gina_k2_p2: boolean;

  begin
    puola1 := lan.eil = p1.eil; puola2 := lan.eil = p2.eil;

    gina_k1_p1 := (k1.eil = lan.eil) and
                  ((p1.stu < k1.stu) and (k1.stu < lan.stu) or
                  (p1.stu > k1.stu) and (k1.stu > lan.stu));

    gina_k1_p2 := (k1.eil = lan.eil) and
                  ((p2.stu < k1.stu) and (k1.stu < lan.stu) or
                  (p2.stu > k1.stu) and (k1.stu > lan.stu));

    gina_k2_p1 := (k2.eil = lan.eil) and
                  ((p1.stu < k2.stu) and (k2.stu < lan.stu) or
                  (p1.stu > k2.stu) and (k2.stu > lan.stu));

    gina_k2_p2 := (k2.eil = lan.eil) and
                  ((p2.stu < k2.stu) and (k2.stu < lan.stu) or
                  (p2.stu > k2.stu) and (k2.stu > lan.stu));

    puolamas := puola1 and not (gina_k1_p1 or gina_k2_p1) or
                puola2 and not (gina_k1_p2 or gina_k2_p2);

  end; { puolamas }
{––––––––––––––––––––––––––––pagrindinė procedūra –––––––––––––––––––}
  procedure raktelio_paieška (b, k1, k2: langelis; 

                              var uodega: integer);

    var galva, iš, i, kont_num: integer;

        len: lenta;

        pasiekta, gali: boolean;

        juda, nejud, lan, bur, buv_lan, buv_juda: langelis;

  begin
    galva := 0; uodega := 0;

    įtraukti_į_eilę (b, k1, k2, 0, uodega); {ši procedūra bus aprašyta vėliau}
    { gal Buratinas pasieks raktelį nestumdydamas konteinerių }
    banga (b, r, k1, k2, len, pasiekta); {ši procedūra bus aprašyta vėliau}
    while not pasiekta do
      begin
        galva := galva + 1;

        paimti_iš_eilės (galva, iš, b, k1, k2); {ši procedūra bus aprašyta vėliau}
        for i := 1 to 4 do { abu konteinerius stumiame visomis kryptimis }
          for kont_num := 1 to 2 do
           if not pasiekta then
              begin
                { stumiamas konteineris yra langelyje juda }
                if kont_num = 1 

                   then begin juda := k1; nejud := k2; end
                   else begin juda := k2; nejud := k1; end;

                { konteineris bus perstumtas į šį langelį }
                lan.eil := juda.eil + dx[i];

                lan.stu := juda.stu + dy[i];

                { Buratinas jį stums iš šio langelio }
                bur.stu := juda.stu - dy[i];

                bur.eil := juda.eil - dx[i];

                { ar Buratinas gali pasiekti langelį iš kurio stums }
                banga (b, bur, k1, k2, len, gali);

                if gali and lentoje (lan) and 

                   not puolamas (juda, lan, nejud)

                   and laisva (lan, k1, k2) then
                   begin { stumiama kol sutinkama kliūtis }
                     while lentoje (lan) and 

                           not puolamas (juda, lan, nejud) and
                           laisva (lan, k1, k2) do
                       begin
                         buv_lan := lan;

                         buv_juda := juda;

                         juda := lan;

                         lan.eil := juda.eil + dx[i];

                         lan.stu := juda.stu + dy[i];

                       end;

                     lan := buv_lan;

                     juda := buv_juda;

                if kont_num = 1

                   then įtraukti_į_eilę (juda, lan, k2, galva, uodega)

                   else įtraukti_į_eilę (juda, k1, lan, galva, uodega);

                { gal iš naujos padėties galima pasiekti raktelį }
                if kont_num = 1

                   then banga (juda, r, lan, k2, len, pasiekta)

                   else banga (juda, r, k1, lan, len, pasiekta)

              end;

            end;

      end;

  end; { raktelio_paieška }
  var b, k1, k2: langelis;

      uodega: integer;

begin
  skaityti (n, b, r, k1, k2, p1, p2); {ši procedūra nebus pateikta}
  raktelio_paieška (b, k1, k2, uodega);

  rezultatai (uodega, Q, sk, ej); {ši procedūra bus aprašyta vėliau}
  spausdinti (sk, ej); {ši procedūra nebus pateikta}
end.

Procedūros skaityti bei spausdinti, kurios perskaito iš bylos pradinius duomenis ir rezultatus surašo į bylą yra labai paprastos, tad jų nepateiksime.

Žemiau pateiksime procedūrą, realizuojančią bangos algoritmą. Ji bangos metodu bando rasti kelią nuo langelio, kuriame yra Buratinas iki raktelio. Konteineriai nestumdomi.

procedure banga (lan1, lan2, k1, k2: langelis;

                 var len: lenta;

                 var pasiekta: boolean);

  { bangos metodu iš langelio lan1 bando pasiekti lan2; konteineriai nestumdomi }
  var i, j, k: integer;

      rastas_naujas: boolean;

      nau: langelis;

begin
  { gal būt tie langeliai sutampa }
  pasiekta := lygu (lan1, lan2);

  if pasiekta then exit;

  for i := 1 to n do
    for j := 1 to n do
      len[i, j] := -1; { langeliai nepasiekti }
  len[lan1.eil, lan1.stu] := 0;

  rastas_naujas := true;

  repeat { kol pasiekiame naujus langelius ir nepasiekėme ieškomojo }
  rastas_naujas := false;

  for i := 1 to n do
    for j := 1 to n do
      if (len[i, j] <> -1)

         then for k := 1 to 4 do
                begin
                  nau.eil := i + dx[k]; nau.stu := j + dy[k];

                  if lentoje (nau) and not puolamas(nau, k1, k2) 

                     and laisva (nau, k1, k2) and not pasiekta

                     and (len[nau.eil, nau.stu] = -1)

                     then begin
                            len[nau.eil, nau.stu] := len[i,j]+1;

                            rastas_naujas := true;

                            pasiekta := lygu (nau, lan2);

                            if pasiekta then exit;

                          end;

                end;

  until not rastas_naujas or pasiekta;

end; { banga }
Procedūros, skirtos darbui su eile yra įprastinės. Šiek tiek skiriasi įtraukimo į eilę procedūra, nes prieš įtraukiant į eilę patikrinama, ar eilėje nėra analogiškos padėties.

procedure įtraukti_į_eilę (b, k1, k2: langelis; iš: integer;

                           var uodega: integer);

  var i: integer;

      rasta: boolean;

      len: lenta;

begin
  { patikrinsime, ar nėra analogiško elemento eilėje }
  i := 1; rasta := false;

  while not rasta and (i < uodega) do
    begin
      if (lygu(Q[i].k1, k1) and lygu(Q[i].k2, k2) or
          lygu(Q[i].k2, k1) and lygu(Q[i].k1, k2))

         then banga (Q[i].b, b, k1, k2, len, rasta);

     i := i + 1;

   end;

  if not rasta

     then begin
            uodega := uodega + 1;

            Q[uodega].b := b;

            Q[uodega].k1 := k1;

            Q[uodega].k2 := k2;

            Q[uodega].iš := iš;

          end;

end; { įtraukti_į_eilę }
procedure paimti_iš_eilės (galva: integer; var iš: integer;

                           var b, k1, k2: langelis);

begin
  iš := Q[galva].iš;

  k1 := Q[galva].k1; k2 := Q[galva].k2; b := Q[galva].b;

end; { paimti_iš_eilės }
Kiek sunkesnės yra procedūros, kurios pagal eilėje esančias būsenas gauna rezultatus. Šiuo atveju rezultatai (t. y. ėjimai) bus gaunami nuo galo. Paskutinysis eilės elementas, kurį rodo uodega, yra tokia padėtis iš kurios Buratinas nebestumdydamas konteinerių gali pasiekti raktelį. Tad bangos metodu iš šios padėties pasiekiamas raktelis ir įsimenama Buratino kelionės pabaiga. Tuomet paimamas kitas eilės elementas: tas iš kurio buvo gauta pastaroji būsena. Kiek reikia pastumiamas konteineris, Buratinas ateina į reikiamą langelį ir veiksmai kartojami kol atrandamas visas kelias.

procedure rasti_ėjimus (lan1, lan2: langelis; len: lenta;

                        var sk: integer; var ėj: ėjimai);

  { Lentoje bangos metodu surašyti ėjimai, padedantys iš pirmojo langelio pasiekti antrąjį; }

  {ši procedūra tuos ėjimus atbula tvarka surašo į masyvą ėj }
  var i, num: integer;

      rasta: boolean;

      nau: langelis;

begin
  while not lygu (lan2, lan1) do
    begin
     i := 0;  rasta := false;

     while (i < 4) and not rasta do
       begin
         i := i + 1;

         num := len[lan2.eil, lan2.stu];

         nau.eil := lan2.eil + dx[i]; 

         nau.stu := lan2.stu + dy[i];

         rasta := lentoje(nau) and (len[nau.eil,nau.stu]+1 = num);

       end;

    sk := sk + 1;

    case i of
      1: ėj[sk] := 'A';

      2: ėj[sk] := 'D';

      3: ėj[sk] := 'V';

      4: ėj[sk] := 'K';

    end;

    lan2 := nau;

  end;

end; { rasti_ėjimus }
procedure rezultatai (uodega: integer; const Q: eilė;

                      var sk: integer; var ėj: ėjimai);

  { pagal eilę suformuojami rezultatai }
  var kiek, i, iš, iš_n: integer;

      sena, nauja, bur, b, k1, k2, b_n, k1_n, k2_n: langelis;

      len: lenta;

      pasiekta: boolean;

begin
  sk := 0; { ėjimų skaičius }
  paimti_iš_eilės (uodega, iš, b, k1, k2);

  banga (b, r, k1, k2, len, pasiekta);

  rasti_ėjimus (b, r, len, sk, ėj);

  while iš <> 0 do
    begin
      paimti_iš_eilės (iš, iš_n, b_n, k1_n, k2_n);

      { pastumiamas konteineris }
      if lygu (k1, k1_n)

         then begin sena := k2; nauja := k2_n; end
         else begin sena := k1; nauja := k1_n; end;

      kiek := abs (sena.eil-nauja.eil) + 

              abs (sena.stu - nauja.stu);

      for i := 1 to kiek do
        begin
          sk := sk + 1;

          if sena.eil < nauja.eil then ej[sk] := 'A'

          else if sena.eil > nauja.eil then ej[sk] := 'V'

          else if sena.stu < nauja.stu then ej[sk] := 'K'

          else ej[sk] := 'D';

       end;

      bur := nauja;

      if sena.eil < nauja.eil

         then bur.eil := bur.eil + 1

      else if sena.eil > nauja.eil then bur.eil := bur.eil - 1

      else if sena.stu < nauja.stu then bur.stu := bur.stu + 1

      else bur.stu := bur.stu - 1;

      { Buratinas paeina į reikiamą langelį }
      banga (b_n, bur, k1_n, k2_n, len, pasiekta);

      rasti_ėjimus (b_n, bur, len, sk, ėj);

      iš := iš_n; b := b_n; k1 := k1_n; k2 := k2_n;

    end;

end; { rezultatai }
Ši procedūra grąžina ėjimus surašytus atbula tvarka. Todėl spausdinant reikės pakeisti tvarką: spausdinti nuo paskutiniojo, o ne nuo pirmojo lentelės ėj elemento.

Žemiau pateiktuose paveiksluose iliustruojami testai.

1 testas. Konteineriai užstoja plėšikus. Raktelis pasiekiamas einant tiesia linija.




2 testas. Raktelį galima pasiekti nestumdant konteinerių. Abu plėšikai vienoje lauko pusėje.




3 testas. Raktelį galima pasiekti nestumdant konteinerių. Abu plėšikai vienoje lauko pusėje.




4 testas. Raktelį galima pasiekti nestumdant konteinerių. Abu plėšikai skirtingose lauko pusėse, tačiau vienoje horizontalėje.




5 testas. Apatinis plėšikas uždengtas, reikia uždengti antrąjį plėšiką. Abu plėšikai vienoje lauko pusėje. Laukas labai mažas.




6 testas. Viršutinis plėšikas uždengtas. Plėšikai skirtingose lauko pusėse.




7 testas. Abu plėšikai vienoje lauko pusėje uždengtas apatinis plėšikas.




8 testas. Abu plėšikai vienoje horizontalėje. Tarp jų stovi vienas konteineris.




9 testas. Neuždengtas nė vienas plėšikas. Abu pasislėpę vienoje horizontalėje. Konteineriai susijungę kampais.




10 testas. Konteineriai stovi gretimuose langeliuose vienas virš kito.




11 testas. Abu plėšikai vienoje horizontalėje. Tarp jų stovi konteineris. Norint pasiekti raktelį, antrą konteinerį būtina pastumti į šoną ir po to į viršų. 




12 testas. Abu konteineriai yra gretimuose langeliuose toje pačioje horizontalėje. Toje pačioje horizontalėje yra ir vienas plėšikas.




13 testas. Abu konteineriai stovi greta toje pačioje horizontalėje. Tarp jų reikia padaryti laisvą vertikalę. Plėšikai pasislėpę aukštesnėse nei konteineriai vertikalėse. Didelis laukas.




14 testas. Abu konteineriai toje pačioje horizontalėje gretimuose langeliuose. Plėšikai pasislėpę vienoje horizontalėje. Tarp konteinerių reikia padaryti laisvą vertikalę. Laukas yra labai mažas.




15 testas. Kairysis konteineris skirtas dešiniajam plėšikui, o dešinysis – kairiajam. Laukas mažas.




16 testas. Didelis laukas. Kairysis konteineris skirtas dešiniajam plėšikui, o dešinysis – kairiajam.




17 testas. Vienas konteineris stovi virš kito konteinerio be tarpų. Plėšikai pasislėpę vienoje lauko pusėje.




18 testas. Didelis laukas. Abu konteineriai stovi vienoje vertikalėje. Abu plėšikai – vienoje horizontalėje. Vienas konteineris yra tarp dviejų plėšikų.




19 testas. Konteineriai stovi vienas virš kito be tarpų. Laukas mažas.




20 testas. Didelis laukas. Abu konteineriai pristumti prie krašto. Plėšikai pasislėpę vienoje horizontalėje.




21 testas. Didelis laukas. Vienas konteineris pristumtas prie krašto.




22 testas. Pats sunkiausias testas. Būtina stumti konteinerį žemyn. Tas pas konteineris panaudojamas uždengti abiem plėšikams.




162. Skaičiai. Po kiekvienos olimpiados išgirstame atsiliepimų: „Nepagalvojau..“, „Tokioj vietoj liko klaida..“, „Pakeisčiau vieną eilutę ir viskas eitų..“.

Visi šie dalyviai buvo sugalvoję neblogą algoritmą tik nemokėjo sudaryti tinkamų testų ir išgaudyti likusių klaidų. 

Kartais girdime ir tokių atsiliepimų: „Žinau kur padariau klaidą, bet „eina“ visi testai!“.

Taigi, tinkamų testų sukūrimas dažnai užima ne mažiau laiko negu paties uždavinio sprendimas.

Šiuo uždaviniu kaip tik ir mokoma kurti testus. Nėra nei gatavų receptų, nei algoritmų kaip tai daryti: reikia stengtis apimti kiek galima daugiau variantų.

Programoje SKAIC padarytas klaidas galima būtų skirstyti į penkias grupes:

1. Ignoruojami žodžiai, kurie nėra kiekiniai skaitvardžiai, pavyzdžiui,

        trylika šimtų –> 13;

        dvidešimt abra kadabra du –> 22.

2. Nereaguojama į tai, kad po „dešimt“ negali būti „vienetų“, pavyzdžiui,

        dešimt vienas –> 11

3. Neatpažįstamas žodis „septyniasdešimt“. Todėl jis ignoruojamas kaip bet kuris ne kiekinis skaitvardis, pavyzdžiui,

        septyniasdešimt vienas –> 1

4. Neatpažįstami (spausdina nulį) kai kurie skaitvardžiai su žodžiais „tūkstantis“, „tūkstančių“, pavyzdžiui,

        dvidešimt vienas tūkstantis du –> 0;

        vienas šimtas tūkstančių –> 0.

5. Klaidingai atpažįstami (spausdina skaičių, nelygų nuliui) kai kurie teisingi ir neteisingi skaitvardžiai su žodžiais „tūkstantis“, „tūkstančiai“, pavyzdžiui,

        dvylika tūkstančiai du –> 12002;

        tūkstantis dvidešimt –> 20.

Visus aukščiau esančius pavyzdžius sujungę į vieną bylą, gausime išsamų testą iliustruojantį visas klaidas. 

Tiesa, norint jas surasti, reikėjo sukurti tikrai ne vieną testą…

163. Geležinkelio grotelės. Idėja gana paprasta: pirmiausia imame aukštį k = 2, iššifruojame tekstą ir jame ieškome rakto. Jei surandame – uždavinys išspręstas, jei ne – padidiname k ir kartojame tuos pačius veiksmus. Taip galų gale rasime mažiausią aukštį.

Kaip dešifruoti tekstą, kai turime kažkokį konkretų aukštį? Panagrinėkime sąlygoje pateiktą pavyzdį (aukštis k = 6). Įsivaizduokime, kad reikia atlikti atvirkštinį darbą: surašyti užšifruoto teksto simbolius atgal į groteles. 

163.2 pav. 

Iššifruoto teksto fragmentas:

	A
	Š
	
	
	
	
	
	
	
	B
	I
	N
	
	
	


Pradžioje pildoma pirmoji eilutė: pirmasis simbolis atsidurs pirmame stulpelyje, antrasis – (2(k – 2) + 1 stulpelyje, trečias – pridėjus prie ankstesniojo dar 2(k – 2. Pastebėsime, kad šie stulpeliai atitinka raidžių vietas tikrame tekste (žr. 163.2 pav.). Kai stulpelio numeris viršija teksto ilgį, reiškia reikia pereiti prie kitos eilutės. Mūsų pavyzdyje pirmoje eilutėje bus įrašytos raidės AI. 

Nuo antrosios eilutės tarpai tarp stulpelių, kuriuose įrašomos raidės, nebebus vienodi. Tolesnė užšifruoto teksto raidė bus rašoma antroje eilutėje, todėl raidė „Š“ atsidurs 2–ame stulpelyje (nes pildoma antra eilutė). Antrojo antrosios eilutės simbolio stulpelį gausime prie 2 (pirmojo antrosios eilutės simbolio stulpelio) pridėję 2(k – 2(e, trečiojo – prie 2–ojo simbolio stulpelio pridėję 2(e–2, ketvirtojo – prie 3–ojo simbolio stulpelio vėl pridėję 2(k–2(e ir t.t. Čia e yra eilutės numeris. Šiuo atveju e = 2. Tas pats sąryšis tinka ir kitoms eilutėms, išskyrus paskutiniąją, kuriai galioja ta pati taisyklė kaip ir pirmajai. Skiriasi tik tuo, kad pirmasis simbolis į k-ąją eilutę rašomas k–ame stulpelyje.

Kadangi tekstas gali būti ilgas (16000 simbolių), tai rakto paieška, kai tiesiog simbolis po simbolio peržiūrimas iššifruotas tekstas, užima nemažai laiko. Be to ją reikės atlikti ne vieną kartą. Galima mėginti paiešką organizuoti kaip nors gudriau. Vienas iš būdų gali būti toks. Suskaičiuojama kiek tekste yra raidžių, kuriomis prasideda raktas, tuomet sudaroma nagrinėjamoji eilutė ir įsidėmima kokiuose stulpeliuose buvo įrašytos pirmosios rakto raidės, jei tai buvo padaryta. Tada kiekvienai tokiai (įstatytai( pirmajai raidei reikia patikrinti, ar toje vietoje tekste dar gali (tilpti( raktas. Jei ne, tai šios pozicijos nebereikės tikrinti. Jei pasirodo, kad raktas jau įrašytas, tai reiškia, kad paiešką galima nutraukti. Jei rakto dar neradome, tai dedame kitą eilutę ir atliekame tuos pačius veiksmus. Jei jau sudėtos visos (pirmosios( raidės, o rakto nebėra kur (įstatyti(, reiškia patekome į aklavietę ir reikia didinti aukštį.

Šis būdas neduoda apčiuopiamo efekto, kai tekstas yra nedidelis, tačiau dideliems tekstams (14–16 tūkstančių simbolių) rezultatai trečdaliu geresni. Dirbant labai greitu kompiuteriu, tas pagerėjimas tėra sekundės dalys…

const MAX_N = 16000; { maksimalus teksto ilgis }
      MAX_K = 6000;  { maksimalus pirmojo simbolio pasikartojimų skaičius }
type tekstas = array [1..MAX_N] of char;

     lentelė = array [1..MAX_K] of integer;

{ žemiau aprašyti globalieji kintamieji, kurių reikšmės nekeičiamos }
var n: integer;   { šifro ilgis }
    šif: tekstas; { pats šifras }
    fr: string;   { fragmentas }
procedure init (var t: tekstas);

  { masyvą t užpildome fiktyviais simboliais }
  var i: integer;

begin
  for i := 1 to n do
    t[i] := chr(1);

end; { init }
procedure dėlioti (e,          { kuri eilutė dėliojama }
                   k: integer; { nagrinėjamas aukštis }
         var poz: integer; { nuo kurios šifro raidės rašoma }
         var t: tekstas;

         var len: lentelė ; { vietų, kur dar galima įrašyti fr, lentelė }
         var ilg,           { tos lentelės ilgis }
             įraš: integer { kelios pirmosios raidės jau įrašytos tekste });

  { sudėlioja e–osios eilutės simbolius }
  var stu, { simbolio, kurį dabar įrašysime, stulpelis }
      post1, post2: integer;

      pirmas: boolean;

begin
  if (e = 1) or (e = k)

     then begin  post1 := 2*k - 2;   post2 := 2*k - 2; end
     else begin  post1 := 2*k - 2*e; post2 := 2*e - 2; end;

  pirmas := true; { pradėsime nuo pirmojo postūmio }
  stu := e; { pirmojo e–sios eil. simbolio stulpelis }
  while stu <= n do
    begin
      t[stu] := sif[poz];

      if sif[poz] = fr[1]  { jei įrašėme pirmą fragmento raidę }
         then begin
               ilg := ilg + 1;

               len[ilg] := stu;  { įsidėmime poziciją }
               įraš := įraš + 1; { gal nuo jos galima įrašyti fragmentą }
              end;

      poz := poz + 1;

      if pirmas then stu := stu + post1

                else stu := stu + post2;

      pirmas := not pirmas;

   end;

end; { dėlioti }
procedure paieška (const t: tekstas; var len: lentelė;

                   var ilg: integer; var rasta: boolean);

  { ieško fragmento fr dalinai iššifruotame tekste t; kartu skaičiuoja

   keliose vietose yra galimybė rasti fragmentą dešifravus tekstą }
  var i, j, stu: integer;

      gerai, tarpas: boolean;

begin
  i := 1;  rasta := false;

  while (i <= ilg) and not rasta do
    begin
      stu := len[i];

      if stu > n - length(fr) + 1 { jei fragmentas nebetilps }
         then begin               { tekste }
                len[i] := len[ilg];

                ilg := ilg - 1;

              end
         else begin { jei fragmentas tilps nuo i–osios pozicijos }
                gerai := true; { nerasta nesutampančių simbolių }
                tarpas := false; { nerasta neįrašytų simbolių }
                j := 0;

                while gerai and (j < length(fr)-1) do
                  begin

                    j := j + 1;

                   { ar galima įrašyti fragmentą pradedant i–a teksto pozicija }
                    if t[stu+j] = chr (1){ jei tekstas neužpildytas }
                       then tarpas := true    

                       else gerai := t[stu+j] = fr[1+j];

                  end;

                 if gerai

                    then if not tarpas

                            then rasta := true

                            else i := i + 1

                    else begin

                           len[i] := len[ilg];

                           ilg := ilg - 1;

                         end;

              end
       end
end; { paieška }
function aukštis: integer;

  var ilg, poz, e, įraš, i, stu, k, pirm: integer;

      rasta: boolean;

      t: tekstas;

      len: lentelė;

begin
  k := 2;     { pradėsime nuo tokio aukščio }
  { suskaičiuojame kiek kartų tekste sutinkama pirmoji fragmento raidė }
  pirm := 0;

  for i := 1 to n do
    if sif[i] = fr[1]

       then pirm := pirm + 1;

  rasta := false;

  while not rasta do
    begin { tikrina, ar tinkamas aukštis k }
      poz := 1;    { pradėsime nuo pirmojo užšifruoto teksto simbolio }
      init (t);    { inicializuojamas masyvas t }
      įraš := 0;   { kelios pirmosios raidės jau įrašytos tekste }
      ilg := 0;    { lentelė len dar tuščia }
      for e := 1 to k do
        begin { sudarome e–ąją eilutę }
          dėlioti (e, k, poz, t, len, ilg, įraš);

          paieška (t, len, ilg, rasta);

          if not rasta

             then if (pirm = įrasš) and (ilg = 0 )

                     { aklavietė – fragmento niekur negalima įrašyti }
                      then break;

        end;

      k := k + 1;

    end;

  aukštis := k - 1;

end; { aukštis }
164. Dvipartinė sistema. Nesunku pastebėti, kad šio uždavinio sprendimui visiškai nesvarbu kaip išsidėstę baltųjų ir juodųjų rajonai. Svarbu tik tai kuriose vietose yra sostinės ir kiek kokių rajonų yra. 

Taigi, pradžioje suskaičiuojame abiejų valstybių plotą ir randame jų sostines.

const MAX = 50; { n ir m maksimumas }
type žemėlapis = array [1..MAX, 1..MAX] of char;

     rajonas = record
                  x, y: integer; { rajono koordinatės }
                end;

procedure dydis_sostinė (žem: žemėlapis; n, m: integer;

                         var balt, juod: integer;

                         var b_sost, j_sost: rajonas);

{ randa abiejų valstybių dydžius ir sostines }
  var x, y: integer;

begin
  balt := 0; juod := 0;

  for x := 1 to n do
    for y := 1 to m do
      begin
        if žem[x, y] in ['.', '+']

           then balt := balt + 1

           else if žem[x, y] in ['*', '-']

                   then juod := juod + 1;

        if žem[x, y] = '+'

           then begin  { rasta baltųjų sostinė }
                  b_sost.x := x; b_sost.y := y

                end
           else if žem[x, y] = '-'

                then begin { rasta juodųjų sostinė }
                       j_sost.x := x; j_sost.y := y

                     end;

       end;

end; { dydis_sostinė }
Tuomet patikriname, ar valstybių teritorijos vientisos. Tai darome iš kiekvienos valstybės sostinės rekursiškai eidami į visas keturias puses. Po to sulyginame tikrąjį kiekvienos valstybės plotą su apeituoju.

164.3 pav.

procedure žemėl (n, m: integer;

                 var gerai, { true, jei teritorijos vientisos }
                 galima: boolean; { true, jei galima pakeisti žemėl. }
                 var žem: žemėlapis);

  { šis kintamasis globalus žemiau esančiai rekursinei procedūrai }
  var kopija: žemėlapis;

  procedure eiti (i, j: integer; simb, spalva: char);

  { simboliu simb pažymi visus nurodytos spalvos rajonus }
  { sujungtus su rajonu (i, j); procedūra rekursinė }

  begin

    kopija[i, j] := simb;

    if (i-1 >= 1) and (kopija[i-1, j] = spalva)

       then eiti (i-1, j, simb, spalva);

    if (i+1 <= n) and (kopija[i+1, j] = spalva)

       then eiti (i+1, j, simb, spalva);

    if (j-1 >= 1) and (kopija[i, j-1] = spalva)

       then eiti (i, j-1, simb, spalva);

    if (j+1 <= m) and (kopija[i, j+1] = spalva)

       then eiti (i, j+1, simb, spalva);

  end; { eiti }
  var balt, juod, b_izol, j_izol, kiek, i: integer;

      b_sost, j_sost, xx, yy: rajonas;

begin
  { rasime abiejų valstybių dydžius ir sostines }
  dydis_sostinė (žem, n, m, balt, juod, b_sost, j_sost);

  { patikrinsime, ar valstybių teritorijos yra vientisos }
  žem[b_sost.x, b_sost.y] := '.';

  žem[j_sost.x, j_sost.y] := '*';

  kopija := žem;

  { pažymime vientisas teritorijas, sujungtas su sostinėmis }
  eiti (b_sost.x, b_sost.y, 'b', '.');

  eiti (j_sost.x, j_sost.y, 'j', '*');

  { patikrinsime, ar yra izoliuotų teritorijų }
  dydis_sostinė (kopija, n, m, b_izol, j_izol, xx, yy);

  if (b_izol = 0) and (j_izol = 0)

     then begin  { teritorijos vientisos, darbas baigiamas }
            gerai := true;

            exit

          end;

...{ Perskirstome rajonus baltiesiems ir juodiesiems }

   ...
end; { keistis }
Jei teritorijos nėra vientisos, tuomet skirstome rajonus valstybėms stengdamiesi, kad nesusidarytų nejungios teritorijos.

Vienas galimų skirstymo būdų būtų toks. Imame bet kurią sostinę (pavyzdžiui, juodųjų) ir nuo sostinės iki kurio nors krašto (kraštas tai viršutinė ar apatinė eilutė, kairysis ar dešinysis stulpelis) „nuvedame“ juodųjų rajonų juostą.

Tačiau kraštą pasirenkame tokį, kad nei jame, nei kelyje nebūtų baltųjų sostinės. 

164. 4 pav.

Nuvedame juostą iki krašto:

	
	
	
	
	
	
	

	
	
	(
	
	
	
	

	
	
	
	
	
	
	

	o
	
	
	
	
	
	


164.5 pav.

Paveikslėliuose juodai užtušuoti juodiesiems priklausantys rajonai, baltai – baltiesiems, pilkai – kol kas nei vienai valstybei nepriskirti rajonai.

O jei turėtume tokią situaciją:

164.6 pav.

juodųjų rajonų juostą reiktų nuvesti taip:

164.7 pav.

Jei juodųjų sostinė yra krašte ir tam krašte nėra baltųjų sostinės, tada nieko nedarome, nes iš vienos pusės kraštą pasiekėme, o iš kitos – rizikuojame padalinti teritoriją į dvi nejungias dalis. 

Pasiekus tinkamą kraštą tolesniam darbui reikės įsiminti kuri eilutė ar stulpelis bus spalvinami juodai (t. y. paskiriami juodųjų valstybei) toliau. Tam sukurtas žemiau esantis tipas. Pavyzdžiui kairė reikš, kad reikės pildyti langelius (rajonus)esančius toje pat eilutėje kaip einamasis langelis, bet kairiau jo.

type linija = (kairė, viršus, dešinė, apačia);

function succ_l (l: linija): linija;

begin
  if l = apačia

     then succ_l := kairė

     else succ_l := succ (l);

end; { succ_l }
procedure spalvinti (nr, nuo, iki, pokytis: integer;

                     spalva: char;

                     eilutė: boolean; { eilutę, ar stulpelį pildyti }
                     var liko: integer; var žem: žemėlapis);

  { eilutės (stulpelio), kurios numeris nr, tušti langeliai nuspalvinami juodai }
  { duota spalva, kai žinoma, kad šia spalva reikia nuspalvinti liko langelių }
  var kiek: integer;

begin
  kiek := abs (nuo - iki) + 1;

  while (kiek > 0) and (liko > 0) do
    begin
      if eilutė and (žem[nr, nuo] = ' ')

         then begin
                žem[nr, nuo] := spalva;

                liko := liko - 1;

              end
         else if not eilutė and (žem[nuo, nr] = ' ')

                 then begin
                        žem[nuo, nr]:= spalva;

                        liko := liko - 1;

                      end;

      nuo := nuo + pokytis;

      kiek := kiek - 1;

    end;

end; { spalvinti }
procedure kraštas (b_sost, j_sost: rajonas; n, m: integer;

                   var juod: integer; var žem: žemėlapis;

                   var l: linija);

  { nuspalvina juodų rajonų liniją nuo sostinės iki to krašto, }
  { kuriame nėra baltųjų sostinės. Tą kraštinę liniją taip pat }
  { paskiria juodiesiems, nes ten tikrai nėra baltųjų sostinės }
begin
  if (b_sost.x <> 1) and (j_sost.x <> n) and
     ((b_sost.y <> j_sost.y) or (b_sost.x > j_sost.x))

     then begin { juodųjų rajonų juosta ateis iki pirmosios eilutės }
            spalvinti (j_sost.y, j_sost.x, 1, -1, '*', false, juod, žem);

            spalvinti (1, j_sost.y, 1, -1, '*', true, juod, žem);

            l := viršus;

          end
  else if (b_sost.x <> n) and (j_sost.x <> 1) and
          ((b_sost.y <> j_sost.y) or (b_sost.x < j_sost.x))

       then begin  { juodųjų rajonų juosta ateis iki paskutiniosios eilutės }
              spalvinti (j_sost.y, j_sost.x, n, 1, '*', false, juod, žem);

              spalvinti (m, j_sost.y, m, 1, '*', true, juod, žem);

              l := apačia;

            end
  else if (b_sost.y <> 1) and (j_sost.y <> m) and
          ((b_sost.x <> j_sost.x) or (b_sost.y > j_sost.y))

       then begin  { juodųjų rajonų juosta ateis iki pirmojo stulpelio }
              spalvinti (j_sost.x, j_sost.y, 1, -1, '*', true, juod, žem);

              spalvinti (n, j_sost.x, n, 1, '*', false, juod, žem);

              l := kairė;

            end
  else if (b_sost.y <> m) and (j_sost.y <> 1) and
          ((b_sost.x <> j_sost.x) or (b_sost.y < j_sost.y))

       then begin { juodųjų rajonų juosta ateis iki paskutiniojo stulpelio }
              spalvinti (j_sost.x, j_sost.y, m, 1, '*', true, juod, žem);

              spalvinti (1, j_sost.x, 1, -1, '*', false, juod, žem);

              l := dešinė;

            end;

end; { kraštas }
Toliau spalviname žemėlapį juodųjų rajonams skirdami po stulpelį (eilutę), pradėdami nuo to krašto kurį pasiekė juodųjų linija. Eilutė (stulpelis) spalvinama juodais langeliais, jei joje nėra baltųjų sostinės. Jei joje yra baltųjų sostinė, pasisukama 90 laipsnių kampu ir spalvinamas arčiausiai krašto esantis nenuspalvintas stulpelis (eilutė).

Pavyzdžiui:

Dabar juodieji jau turi reikiamą skaičių rajonų. Likusius skiriame baltiesiems.

Spalvinant žemėlapį svarbu tinkamai pasirinkti stulpelius (eilutes), kuriuos pildysime. Žemiau parodyta situacija, kai netinkamai pasirinkus spalvinamą stulpelį, nenuspalvinta teritorija tapo nejungi:

Šiuo atveju nuspalvinus viršutinę eilutę (164.9 e) pav.), reikėjo pasisukti ne į kairę o į dešinę. 

procedure sudaryti (n, m, { žemėlapio matmenys }
                    balt, juod: integer; { abiejų valstybių dydžiai }
                    b_sost, j_sost: rajonas;

                    var žem: žemėlapis { pakeistasis žemėlapis });

  { sudaro naują žemėlapį }

var viso, eil_v, eil_a, st_d, st_k, i, j: integer;

      pavojinga, l: linija;

      galima: boolean;

begin
  { žemėlapis išvalomas }
  for i := 1 to n do
    for j := 1 to m do
      žem[i, j] := ' ';

  { pažymimos sostinės }
  žem[b_sost.x, b_sost.y] := '.';

  žem[j_sost.x, j_sost.y] := '*';

  balt := balt - 1; juod := juod - 1;

  { juodiesiems padaromas išėjimas nuo sostinės iki krašto }
  { stengiantis, kad baltųjų sostinė nebūtų tame pačiame krašte }
  kraštas (b_sost, j_sost, n, m, juod, žem, l);

  eil_v := 1; eil_a := n;

  st_k := 1;  st_d := m;

  pavojinga := succ_l(succ_l(l));

  while juod > 0 do
    begin
      case l of
        kairė: begin
                 galima := (b_sost.y <> st_k);

                 if galima then
                   begin
                    if žem[1, st_k] = '*'

                      then spalvinti (st_k, 1, n, 1,  '*', false, juod, žem)

                      else spalvinti (st_k, n, 1, -1, '*', false, juod, žem);

                    st_k := st_k + 1;

                   end
          end;

        dešinė: begin
                  galima := (b_sost.y <> st_d);

                  if galima then
                   begin
                     if žem[1, st_d] = '*'

                      then spalvinti (st_d, 1, n, 1, '*', false, juod, žem)

                      else spalvinti (st_d, n, 1, -1, '*', false, juod, žem);

                      st_d := st_d - 1;

                    end
                end;

        viršus: begin
                  galima := (b_sost.x <> eil_v);

                  if galima then
                    begin
                      if žem[eil_v, 1] = '*'

                        then spalvinti (eil_v, 1, m, 1, '*', true, juod, žem)

                        else spalvinti (eil_v, m, 1, -1, '*', true, juod, žem);

                      eil_v := eil_v + 1;

                    end
                end;

        apačia: begin
                  galima := (b_sost.x <> eil_a);

                  if galima then
                    begin
                      if žem[eil_a, 1] = '*'

                       then spalvinti (eil_a, 1, m, 1, '*', true, juod, žem)

                       else spalvinti (eil_a, m, 1, -1, '*', true, juod, žem);

                      eil_a := eil_a + 1;

                    end
                end;

      end; { case }

      { apskaičiuosime keliomis kryptimis galime eiti }
      { jei nuspalvinta iš viso viena linija, galima spalvinti }
      { linijas iš abiejų jos šonų, jei nuspalvinta daugiau linijų, }
      { galima spalvinti bet kurią (jei ten nėra baltųjų sostinės ir }
      { jei ten nėra pavojinga (164.9 f) pav. }
      viso := ord(b_sost.y <> st_d) + ord(b_sost.y <> st_k) +

              ord(b_sost.x <> eil_a) + ord(b_sost.x <> eil_v);

      if (viso = 1) or (succ_l(l) <> pavojinga)

         { jei nepavojinga pildyti tolesnę liniją arba nėra pasirinkimo }
         then l := succ_l(l)

         else l := succ_l(succ_l(l));

    end;

  { užpildome likusius langelius }
  for i := 1 to n do
    for j := 1 to m do
      if (žem[i, j] = ' ')

          then if balt > 0

                  then žem[i, j] := '.'

                  else žem[i, j] := '*'

end; { sudaryti }
Ši procedūra nenumato vieno atvejo, kai gali nepasisekti padalinti rajonų. Taip gali atsitikti jei n ( 1 ar m ( 1.

Taigi, baigus spalvinimą būtinai dar kartą reikia patikrinti, ar valstybių teritorijos vientisos. Dabar jau nesunku pabaigti procedūra žemel.
procedure žemėl (n, m: integer;

                 var gerai, { true, jei teritorijos vientisos }
                 galima: boolean; { true, jei galima pakeisti žemėl. }
                 var žem: žemėlapis);

  var balt, juod, b_izol, j_izol, kiek, i: integer;

      b_sost, j_sost, xx, yy: rajonas;

      kopija: žemėlapis;

begin
  { rasime abiejų valstybių dydžius ir sostines }
  dydis_sostinė (žem, n, m, balt, juod, b_sost, j_sost);

  { patikrinsime, ar valstybių teritorijos yra vientisos }
  žem[b_sost.x, b_sost.y] := '.';

  žem[j_sost.x, j_sost.y] := '*';

  kopija := žem;

  { pažymime vientisas teritorijas, sujungtas su sostinėmis }
  eiti (b_sost.x, b_sost.y, 'b', '.');

  eiti (j_sost.x, j_sost.y, 'j', '*');

  { patikrinsime, ar yra izoliuotų teritorijų }
  dydis_sostinė (kopija, n, m, b_izol, j_izol, xx, yy);

  if (b_izol = 0) and (j_izol = 0)

     then begin  { teritorijos vientisos, darbas baigiamas }
            gerai := true;

            exit

          end;

...{ Perskirstome rajonus baltiesiems ir juodiesiems }

  sudaryti (n, m, balt, juod, b_sost, j_sost, žem);

  { tikinsime, ar pavyko sudaryti }
  if (n = 1) or (m = 1)

     then begin
            kopija := žem;

            { pažymime vientisas teritorijas, sujungtas su sostinėmis }
            eiti (b_sost.x, b_sost.y, 'b', '.');

            eiti (j_sost.x, j_sost.y, 'j', '*');

            { patikrinsime, ar yra izoliuotų teritorijų }
             dydis_sostinė (kopija, n, m, b_izol, j_izol, xx, yy);

             galima := (j_izol = 0) and (b_izol = 0);

          end
      else galima := true;

end; { keistis }

165. Kam lygi suma. Pabandykime išsiaiškinti kaip atmintyje saugomas slankiojo kablelio realusis skaičius.

Kai spausdiname realųjį skaičių, jis atspausdinamas dešimtaine sistema. Pavyzdžiui: –1,23456789E–100. Realiojo skaičiaus dalis prieš raidę „E“ mantise, skaičiaus dalis po raidės „E“ – eile. 




Pats realusis skaičius lygus: skaičius = mantisė(10eilė. Praktiškai eilė parodo per kiek skaitmenų ir į kurią pusę mantisėje reikia paslinkti kablelį. Pavyzdžiui:

1.458E6 ( 1,458(106( 1458000.0

Tačiau kompiuterio atmintyje realiojo skaičiaus mantisė ir eilė saugomos dvejetainiu pavidalu. T.y. visi skaitmenys yra tik nuliukai arba vienetukai. Pavyzdžiui:




Šį kartą raidę „E“ parašėme tik aiškumo dėlei. Mat ir realiojo skaičiaus mantisei ir realiojo skaičiaus eilei saugoti atmintyje skiriamas fiksuotas bitų skaičius. Taigi nereikia jokios „E“ raidės eilei nuo mantisės atskirti. Taip pat nereikia įsidėmėti kurioje vietoje mantisėje padėtas kablelis, nes laikoma, kad jis yra po pirmojo mantisės skaitmens. O pirmasis mantisės skaitmuo niekada nelygus nuliui. Vienintelė išimtis – kai pats realusis skaičius lygus nuliui. 

Toliau šiame uždavinyje nagrinėsime ir visas operacijas atliksime tik su dvejetainiais skaičiais.

Realusis skaičius dvejetine sistema lygus: skaičius = mantisė(2eilė. Ir šį kartą eilė parodo per kelis skaitmenis reikia paslinkti kablelį:

1,1011001E101 ( 1,1011001(2101 ( 110110,012
Šiuo atveju paslenkama per penkis skaitmenis, nes 1012 ( 510.

Išsiaiškinę kaip atmintyje saugomas realusis skaičius, sukonstruosime situaciją, kurioje a+1.0=a. Prieš sudedant du realiuosius skaičius (a ir 1.0) būtina patikrinti, ar sutampa jų eilės. Jei ne, tai prieš sudedant būtina suvienodinti jų eilę. Rezultato eile pasirenkama didesnioji iš dėmenų eilių. 

Pavyzdžiui, a ( 1,000111E101. a eilė lygi 101. Skaičiaus 1 eilė lygi nuliui: 1,0 ( 1,0(20. Taigi, reikia padidinti vieneto eilę:

1,0(20 ( 0,00001(2101.

Jau galime a padidinti vienetu: 1,000111E101 + 0,000010E101 ( 1,001001E101.

Pažiūrėkime, kas gausis, jei a eilė bus didesnė, nei maksimalus dvejetainės mantisės ilgis . Tarkime, mantisę gali sudaryti ne daugiau aštuonių skaitmenų. O skaičiaus a eilė lygi 910 ( 10012. Pavyzdžiui, a (1,00001111E1001. Suvienodinant eiles mažesniajame dėmenyje (vienete) teks pastumti kablelį per 9 skaitmenis:

1,0(20 ( 0,000000001(21001. Kadangi mantisė gali saugoti tik aštuonis skaitmenis, vienetas pavirs nuliu: 0,0000000(21001(0,0. Gausime, kad suma bus lygi pradiniam skaičiui a.

Sąlyga prašo rasti mažiausią tokį skaičių. Todėl mantisė bus lygi 1,000.... Nes tai mažiausia teigiama nenulinė mantisė. O tai reiškia, kad mūsų ieškomas skaičius lygus dvejeto laipsniui.

Žemiau parašyta programėlė padės jį surasti. 

program skaičiaus_paieška;

  {$N+}

  var a: double;

begin
  a := 1;

  repeat
    a := 2 * a;

  until a + 1 = a;

  writeln (a:24:1);

end.

Ieškomasis skaičius lygus 1,84467440737095516E+19. Jei skaičiaus ieškotumėte eksperimentuodami su įvairiais realiais skaičiais, gautumėte atsakymą 1,84467440737095506E+19. Šie abu skaičiai dvejetainiu pavidalu atmintyje vaizduojami vienodai. Skirtumas atsiranda juos užrašant dešimtainiu pavidalu.

Vyresniųjų grupė

166. Reiškinys. Uždavinio sąlyga turi rimtą kabliuką. Sakykime, duotas toks reiškinys: a+b–c+d. Jei pirma lygiagrečiai sudėtume, o po to atimtume, gautume klaidingą rezultatą: a+b–c+d ( (a+b)–(c+d). Galima skaičiuoti kitaip: a+b–c+d ( (a+b)+(d–c). 

Sąlygoje pasakyta, kad draudžiama atskliausti ir prastinti, tačiau aiškiai nesuformuluota, ar galima kitaip matematiškai pertvarkyti veiksmus. Neaišku, ar leistina keisti kintamuosius vietomis. Susitarsime, kad tai leidžiama daryti.

Taip pat jei sutiksime reiškinį a*b/c*d, laikysime, kad jo reikšmę galime apskaičiuoti per du laiko vienetus:

1a :( a(b; 1b :( d/c; 2a :( 1a(1b;

Ieškant minimalaus laiko vienetų skaičiaus, sudėties operacija yra identiška atimties operacijai, o daugybos operacija – dalybai. Be abejo, jei reiktų rasti pačią reiškinio reikšmę, taip nebūtų. Kintamųjų skaičius bei vardai taip pat neturi jokios įtakos uždavinio sprendimui.

Tad pirmiausia visas dalybos bei atimties operacijas pakeiskime daugybos bei sudėties operacijomis, o visus kintamųjų vardus pakeiskime į a. Algoritmas yra gan paprastas: atliekame visus veiksmus, kuriuos galime atlikti pirmu laiko momentu. Po to atliekame visus veiksmus, kuriuos galime atlikti antru laiko momentu. Ir taip kartojame kol atliekami visi reikalingi veiksmai. Kiekvienoje sudėties ar daugybos operacijoje dalyvauja du kintamieji, kurie yra sudedami arba sudauginami. Atlikus operaciją, tos operacijos rezultatą saugosime pirmajame kintamajame, o antrąjį kintamąjį bei pačią operaciją galime panaikinti (ištrinti iš simbolių eilutės). Imkime sąlygoje pateiktą pavyzdį. Kad pavyzdys būtų aiškesnis, nevienodinsime nei kintamųjų vardų nei operacijų: 

a+b+(c+d)*(e+f)–(c+d)*e;

Per pirmą laiko vienetą galime atlikti šiuos keturis veiksmus:

a :( a + b; c :( c + d; e :( e + f; c :( c + d;

Pašaliname nereikalingus kintamuosius bei operacijas:

a +(c)*(e)–(c)*e;

Belieka pašalinti nereikalingus skliaustus, gaubiančius vieną kintamąjį:

a +c*e–c*e;

Antru laiko momentu galime atlikti du veiksmus:

c :( c*e; c :( c*e;

Vėl pašalinę nereikalingus kintamuosius gauname: a +c–c. Trečiu laiko momentu atlikę sudėtį ir ketvirtu laiko momentu – atimtį gausime vieną kintamąjį – reiškinio reikšmę. Taigi, mums užteko keturių operacijų. 

Loginėje lentelėje reik įsiminsime, kuriuos reiškinio simbolius (kintamųjų vardus bei operacijas) atlikus visus įmanomus veiksmus vienu laiko momentu reikės pašalinti.

const MAX = 255; { maksimalus reiškinio ilgis }
type sk_lent = array [1..MAX] of integer;

     log_lent = array [1..MAX] of boolean;

procedure keisti (var reiš: string);

  { kintamųjų vardus pakeičia į a, atimtį – į sudėtį, dalybą – į daugybą }
  var i: integer;

begin
  for i := 1 to length(reiš) do
    if reiš[i] in ['a'..'z']

       then reiš[i] := 'a'

       else if reiš[i] = '-'

               then reiš[i] := '+'

               else if reiš[i] = '/'

                       then reiš[i] := '*';

end; { keisti }
function skaičiuoti (reisš: string): integer;

  var num, kiek, i: integer;

      reik: log_lent;

begin
  keisti (reiš); { suvienodinami kintamųjų vardai }
  kiek := 0;     { laiko vienetų skaičius }
  while reiš <> 'a' do
    begin
      for i := 1 to length(reiš) do
        reik[i] := true; { nė vieno reiškinio simbolio nereikia pašalinti }
      { atliekame visas operacijas, kurias galime atlikti per vieną laiko momentą}
      atlikti_oper (reiš, reik); 

      num := 1;      { pašaliname nebereikalingus reiškinio simbolius }
      for i := 1 to length(reiš) do
        if not reik[i]  then delete(reiš, num, 1)

                        else num := num + 1;

      { išmetame nereikalingus skliaustus }
      for i := 1 to length(reiš)-2 do

        if (reiš[i] = '(') and (reiš[i+2] = ')')

           then begin

                  delete (reiš, i+2, 1);

                  delete (reiš, i, 1);

                end;

        kiek := kiek + 1;

    end;

  skaičiuoti := kiek;

end; { skaičiuoti }
Liko išsiaiškinti kuriuos veiksmus galima atlikti per vieną laiko momentą. Jei reiškinyje nebūtų nei daugybos operacijų, nei skliaustų, tai iš eilės galėtume atlikti visus veiksmus. Deja, taip nėra. Pradžioje tarkime, kad reiškinyje nėra daugybos operacijų. T. y. tik skliaustai riboja veiksmų atlikimo tvarką. Imkime tokį pavyzdį: 

a+b+c+(d+e)+(f+g)+h+(i+(j+(k+l)+m)+n)+e;

Kiekvienam reiškinio simboliui apskaičiuosime jo gylį – t.y. keli skliaustai jį gaubia. Gausime tokią lentelę. Pirmoje eilutėje įrašytas reiškinio simbolis, antroje – jo gylis. 

Reiškinys     a+b+c+(d+e)+(f+g)+h+(i+(j+(k+l)+m)+n)+e

Gylis        000000111110111110001112223333322211100

Operacijas atliekame iš kairės į dešinę. Pradedame nuo pirmojo kintamojo a. Jo gylis lygus nuliui. Reiškinyje ieškome kito kintamojo, kurio gylis taip pat būtų lygus nuliui. Tai kintamasis b. Juos sudedame: a :( a + b. Įsidėmime, kad kintamieji a ir b jau panaudoti, o kintamasis b – nebereikalingas. Imame tolesnį kintamąjį c. Jo gylis taip pat lygus nuliui. Ieškome kito nepanaudoto kintamojo, kurio gylis būtų toks pat. Tai bus kintamasis h. Vėl įsidėmime panaudotų bei nebereikalingo kintamųjų vardus. 

Imame tolesnį nepanaudotą kintamąjį d. Jo gylis jau lygus vienam. Reiškinyje ieškome kintamojo, kurio gylis sutaptų su kintamojo d gyliu. Tai bus kintamasis e. Jei kartais kintamajam poros nerastume, jį praleidžiame ir imame tolesnį kintamąjį. Ir taip kartojame veiksmus, kol atliekame visas reikalingas operacijas. 

Atkreipkite dėmesį, kad gali būti ir tokia situacija: (a+b+c)+(d+e+f). Šiuo atveju pirmiausia sudėsime a ir b. Kintamųjų c bei d gyliai lygūs vienam. Tačiau jų dar negalima sudėti. Nuo to „apsaugo“ tarp skliaustų esantis pliusas – mat jo gylis lygus nuliui. Taigi, patiksliname: du to paties gylio kintamuosius galima sudėti tik tada, jei tarp jų nėra nė vieno simbolio, kurio gylis būtų mažesnis, nei jų pačių. Simbolių, kurių dydis gilesnis, gali būti. 

Minėto reiškinio a+b+c+(d+e)+(f+g)+h+(i+(j+(k+l)+m)+n)+e reikšmė būtų apskaičiuojama taip:

Pirmas laiko momentas: 

a :( a+b; c :(c+h; d :( d+e; f :( f+g; i :( i+n; j :( j+m; k :( k+l;
Gauname reiškinį: a+c+d+f+(i+(j+k))+e;

Antras laiko momentas:

a :( a+c; d :(d+f; j :( j+k;
Gauname reiškinį: a +d +(i+j)+e;

Trečias laiko momentas:

a :( a +d; i :( i+j;

Gauname reiškinį: a +i+e;

Likusio reiškinio reikšmę galima suskaičiuoti per ketvirtą–penktą laiko momentus. 

Dabar prisiminkime, kad reiškinyje gali būti ir daugybos operacijų. Kai kurias sudėties operacijas buvusiame pavyzdyje pakeiskime daugybos operacijomis. Gausime pavyzdžiui tokį reiškinį: a+b*c+(d+e)*(f+g)*h+(i+(j*(k*l)+m)*n)*e;

Dabar jau nebegalime sudėti kintamųjų a ir b: pirma teks atlikti daugybą. Tam įvesime blokus. Operaciją su dviem kintamaisiais galėsime atlikti tik tada, jei jie priklausys tam pačiam blokui.

Pradžioje kiekvieną reiškinio simbolį paskirsime į pirmąjį bloką. Tuomet reiškinyje ieškome daugybos operacijų. Radę daugybos operaciją jos bloko numerį padidinkime vienetu. Tuomet einame į kairę nuo šio simbolio kol 

· prieiname reiškinio pradžią arba 

· pasiekiame simbolį, kurio gylis mažesnis nei daugybos operacijos simbolio gylis arba
· pasiekiame sudėties operaciją, kurios gylis lygus daugybos operacijos simbolio gyliui;

Visų pakely sutiktų simbolių blokus padidiname vienetu. Tą patį padarome ir eidami į dešinę. 

Reiškinys     a+b*c+(d+e)*(f+g)*h+(i+(j*(k*l)*m)*n)*e

Gylis        000000111110111110001112223333322211100

Blokas       112221333333333333312225556666655533322

Kiekvieną reiškinio simbolį paskyrę į atitinkamą bloką bei suskaičiavę kiekvieno simbolio gylį jau galime vienareikšmiškai pasakyti kurias operacijas galima atlikti per vieną laiko momentą. Taigi, operaciją su dviem kintamaisiais galima atlikti tik tada, jei:

· abiejų kintamųjų gyliai sutampa ir abu priklauso tam pačiam blokui;

· tarp šių kintamųjų nėra nė vieno simbolio, kurio gylis būtų mažesnis nei tų kintamųjų gylis;

· tarp šių kintamųjų nėra nė vieno simbolio, kuris priklausytų blokui su mažesniu numeriu, nei šių kintamųjų bloko numeris.

procedure atlikti_oper (var reiš: string; 

                        var naud: log_lent);

  { atlieka visas operacijas, kurias galima atlikti per vieną laiko momentą }
  procedure rasti_gylį (reiš: string; var gylis: sk_lent);

    { randa kiekvieno simbolio gylį }
    var i, gyl: integer;

  begin
    gyl := 1;  { pirmojo simbolio gylis }
    for i := 1 to length (reiš) do
      begin
        if reiš[i] = '(' then gyl := gyl + 1;

        gylis[i] := gyl;

        if reiš[i] = ')' then gyl := gyl - 1;

      end;

  end; { rasti_gylį }
  function ribose (reiš: string; x: integer): boolean;

  begin
    ribose := (1 <= x) and (x <= length(reiš))

  end; { ribose }
  procedure paeiti (nuo, gyl: integer; 

                    gylis: sk_lent; reiš: string;

                    pirmyn: boolean; var blokas: sk_lent);

  begin
    while ribose(reiš, nuo) and (gylis[nuo] >= gyl) and
          ((reiš[nuo] <> '+') or (gylis[nuo] > gyl)) do
      begin
        blokas[nuo] := blokas[nuo] + 1;

        if pirmyn

           then nuo := nuo + 1

           else nuo := nuo - 1;

      end;

  end; { paeiti }
  procedure skaidyti_blokais (gylis: sk_lent; reiš: string; 

                              var blokas: sk_lent);

    { išskaido reiškinį į blokus }
    var i: integer;

  begin
    for i := 1 to length(reiš) do
      blokas[i] := 1;

    for i := 2 to length(reiš)-1 do
     if reiš[i] = '*'

       then begin
             blokas[i] := blokas[i] + 1;

             paeiti (i-1, gylis[i], gylis, reiš, false, blokas);

             paeiti (i+1, gylis[i], gylis, reiš, true, blokas);

            end; { if }
  end; { skaidyti_blokais }
  var gylis, blokas: sk_lent;

       i, iki: integer;

       rasta, tas_pats_blokas: boolean;

  begin { atlikti_veiksmus }
    rasti_gylį (reiš, gylis);

    skaidyti_blokais (gylis, reiš, blokas);

    i := 1;

    while i <= length(reiš)-1 do
      if (reiš[i] = 'a') and naud[i]

       then begin
             { ieškome artimiausio kintamojo, kurio gylis ir blokas sutaptų }
             rasta := false; tas_pats_blokas := true;

             iki := i;

             while (iki+1 <= length(reiš)) and 

                   tas_pats_blokas and not rasta and 

                   (gylis[i] <= gylis[iki]) do
               begin
                 iki := iki + 1;

                 tas_pats_blokas := blokas[iki] >= 

                                    blokas[i];

                 rasta := (gylis[i] = gylis[iki]) and 

                          tas_pats_blokas and
                          (blokas[i] = blokas[iki]) and 

                          (reiš[iki] = 'a') and naud[iki];

               end;

              if rasta then begin
                              naud[iki-1] := false;

                              naud[iki] := false;

                           end;

                i := i + 1;

            end
       else i := i + 1;

end; { atlikti_oper }
167. Žodžių ratas. Šis uždavinys sprendžiamas grįžimo metodu, t. y. perrenkant visus įmanomus variantus. Perrinkimas atliekamas taip. Sudaroma dvimatė žodžių galūnių lentelė g. Šios lentelės elementas g[i, j] pasako, ar i–ojo žodžio galas sutampa su j–ojo žodžio pradžia. Žodžiai saugomi įrašų lentelėje. Joje įsimenamas ne tik pats žodis, bet ir jo pirmosios raidės vieta rate.

Pradėjus perrinkimą fiksuojamas pirmasis žodis ir jis įrašomas į ratą pradedant pirmuoju rato langeliu. Toliau rekursiškai bandomi dėti visi kiti žodžiai. Užpildžius ratą suskaičiuojama į jį įrašytų žodžių sumą ir patikrinama, ar negautas geresnis sprendinys. 

Išbandžius visus atvejus, kai rate įrašytas pirmasis žodis, fiksuojamas antrasis žodis ir įrašomas į ratą pradedant pirmuoju simboliu. Po to į ratą rekursiškai bandoma įrašyti žodžius nuo trečiojo iki paskutiniojo. Ir t. t.

Uždavinys turi nemažą „kabliukų“ į kuriuos reiktų atsižvelgti. Juos išvardinome žemiau.

– Ratą gali sudaryti ir vienas žodis.

– Jei ratą sudaro vienas žodis, jo pradžia nebūtinai turi persikloti su jo pabaiga.

– Gali būti žodžių, kurie yra kitų žodžių vidinės dalys (pvz. RATAS ir ATA). Todėl užpildžius ratą dar kartą peržiūrimi visi į ratą neįtraukti žodžiai ir bandoma juos įtraukti.

– Gali būti žodžių, kurie yra kitų žodžių pradžios ar pabaigos. Tokias situacijas reikia numatyti rašant algoritmą. Taip pat gali atsitikti kad ieškomas sprendinys bus gautas tik tada, kai tie žodžiai bus įrašyti skirtingose rato vietose.

– Žodžiai gali persikloti skirtingu raidžių skaičiumi (pvz. žodžiai SASAS ir SAS persikloja dviem būdais: SASASAS ir SASAS). Todėl negalima nagrinėti tik ilgiausio persidengimo.

– Tas pats žodis žodyne gali kartotis kelis kartus.

const MAX = 20;    { maksimalus žodžių skaičius }
      ILGIS = 20;  { maksimalus žodžio ilgis }
type žodžiai = array [1..MAX] of
                 record
                   žod: string[ILGIS]; { pats žodis }
                   poz_r,          { jo pozicija rate }
                   poz_s: integer; { ir konstruojamame sprendinyje }
                   dalis: boolean; { ar įtrauktas į ratą kaip kito žodžio dalis }
                 end;

     kv_lent = array [1..MAX, 1..MAX] of set of 0..ILGIS;

{ žemiau aprašyti globalieji kintamieji }
var n,               { pozicijų skaičius rate }
    žod_sk: integer; { duotų žodžių skaičius }
    ž: žodžiai;      { duotieji žodžiai }
procedure vienas_žodis (var maks: integer);

{ išnagrinėja atskirą atvejį, kai ratą sudaro vienas nepersiklojantis }
{ žodis. Žinoma, kad iki šiol nei vieno sprendinio nerasta }
  var i: integer;

begin
  i := 0; maks := 0;

  repeat
    i := i + 1;

    if length(ž[i].žod) = n

       then begin
              ž[i].poz_r := 1;

              maks := n;

            end;

  until (maks <> 0) or (i > n)

end; { vienas_žodis }
procedure galūnės (var g: kv_lent);

{ užpildo masyvą g. Jei s priklauso g[i, j], tai reiškia, kad }
{ s paskutiniųjų i–ojo žodžio raidžių sutampa su s pirmų j–ojo žodžio raidžių }
  var i, j, k, viso: integer;

      pir, ant: string;

begin
  for i := 1 to zod_sk do
    for j := 1 to zod_sk do
      begin
        g[i, j] := [];

        k := 1; { pradžioje imsime tokio ilgio galūnę }
        while (k <= length(ž[i].žod)) and 

              (k <= length(ž[j].žod)) do
              { ar sutaps ilgio k galas su pradžia }
         begin
           { atskirsime k ilgio galūnę bei pradžią }
           pir := copy (ž[i].žod, length(ž[i].žod)-k+1, k);

           ant := copy (ž[j].žod, 1, k);

           if pir = ant

              then g[i, j] := g[i, j] + [k];

           k := k + 1;

         end;

      end;

end; { galūnės }
procedure dalys (var suma: integer);

{ į užbaigtą ratą įtraukią žodžius, kurie yra kitų žodžių dalys }
  var i, j: integer;

begin
  for i := 1 to žod_sk do
    if ž[i].poz_s > 0

       then for j := 1 to zod_sk do
             if (ž[j].poz_s = 0) and 

                (pos(ž[j].žod, ž[i].žod) > 0) and (i <> j)

                { jei j–as žodis nėra rate ir yra i–tojo žodžio dalis }
             then begin
                    ž[j].poz_s := ž[i].poz_s + 

                                  pos(ž[j].žod, ž[i].žod) - 1;

                    suma := suma + length(ž[j].žod);

                    ž[j].dalis := true

                  end;

end; { dalys }
procedure dėlioti (var galima: boolean; { ar galima sudaryti ratą }
                   var viso,   { kiek žodžių jame tilps }
                   maks: integer { žod. ilgių suma geriausiame sprendinyje });

  { procedūra, gaubianti rekursinę procedūrą }
  { žemiau aprašyti kintamieji globalūs rekursinei procedūrai }
  var g: kv_lent;    { g[i, j] rodo kelios i–tojo žodžio pabaigos }
                     { raidės sutampa su j–tojo žodžio pradžia }
      pirm: integer; { žodžio, kuriuo pradedamas ratas, numeris }
  procedure dėti (i, { įrašomo žodžio indeksas }
                  r,   { raidžių, sutampančiu su ankstesniu žodžiu, skaičius }
                  užim, { užimtų pozicijų skaičius }
                  suma: integer { žodžių ilgių suma });

  { rekursinė procedūra dėti į ratą įrašo i–ąjį žodį }
    var ilgumas, k, j: integer;

  begin
    { įrašome i–ąjį žodį; sutampa pirmosios r raidės }
    ilgumas := length(ž[i].žod); { i–ojo žodžio ilgis }
    suma := suma + ilgumas;

    ž[i].poz_s := užim - r + 1;

    užim := užim + ilgumas - r;

    { jei dėliojimas baigtas }
    if užim > n

       then begin
              if (užim-n) in g[i, pirm] { jei ratas užsidaro } then
                begin
                  { prijungiame žodžius, kurie yra kitų žodžių dalys }
                  dalys (suma);

                  { ar radome geresnį sprendinį }
                  if suma > maks

                     then begin
                            maks := suma;

                            for k := 1 to žod_sk do
                              ž[k].poz_r := ž[k].poz_s;

                          end;

                  { grąžiname reikšmes – iš rato pašaliname žodžius, }
                  { kurie yra kitų žodžių dalys }
                  for k := 1 to zod_sk do
                    if ž[k].dalis

                       then begin
                              ž[k].dalis := false;

                              ž[k].poz_s := 0;

                            end;

                end { jei ratas užsidaro }
            end { jei dėliojimas baigtas }
       else { dėliojame toliau }
         for j := pirm+1 to žod_sk do { pirmuosius žodžius praleidžiame }
           if (ž[j].poz_s = 0) and (g[i, j] <> [])

              { jei j–as žodis nėra rate ir jį galima jungti prie i–tojo }
              then for k := 1 to length(ž[j].žod) do
                     if k in g[i, j]

                        then dėti (j, k, užim, suma);

    { atstatome reikšmes }
    ž[i].poz_s := 0;

    ž[i].dalis := false;

  end; { dėti }
  var i, užim, suma: integer;

begin
  galūnės (g);

  for i := 1 to žod_sk do
    with ž[i] do
      begin
        poz_r := 0; { žodis neįtrauktas į ratą }
        poz_s := 0; { nei į konstruojamą sprendinį }
        dalis := false;

      end;

  maks := 0;  { maksimalus sutalpintų raidžių skaičius optimaliame sprendinyje}
  užim := 0;  { neužimta nei viena rato pozicija }
  suma := 0;

  { patikrinsime, ar ratą gali sudaryti vienas nepersiklojantis žodis }
  if n <= ILGIS

     then vienas_žodis (maks);

     { ieškosime kitų sprendinių }
  for pirm := 1 to zod_sk do  { ratą pradėsime i–tuoju žodžiu }
    dėti (pirm, 0, užim, suma);

    { i–tasis žodis bus rašomas nuo pirmosios pozicijos, nėra raidžių, }
    { sutapusių su prieš tai buvusiu žodžiu }
  { baigę dėlioti žodžius rasime kiek žodžių tilpo į ratą }
  viso := 0;

  for i := 1 to žod_sk do
    if ž[i].poz_r <> 0

       then viso := viso + 1;

  galima := viso <> 0;

end; { dėlioti }
Įvertinsime algoritmo vykdymo laiko eilę. Plačiau apie šią sąvoką galite pasiskaityti prie 159 uždavinio sprendimo. Rekursinių perrinkimo algoritmų vykdymo laikas auga eksponentiškai. Bandysime įvertinti algoritmo vykdymo laiką pakankamai blogu atveju.

Žodyne yra m žodžių. Pirmąjį žodį galime parinkti m būdų. Atrodytų, kad antrąjį žodį galime parinkti daugiausia (m–1) būdu. Tačiau nepamirškite, kad vieno žodžio pabaiga su kito žodžio pradžia gali sutapti nebūtinai vienoje vietoje (prisiminkime SASAS ir SAS). Taigi, variantų skaičius gali būti žymiai didesnis. Sąlygoje įvestas ribojimas, kad sąryšių skaičius žodyne neviršija 110. 

Bandykime surasti pakankamai blogą atvejį. Toks atvejis būtų, jei žodyne būtų daug žodžių (pvz. 20), sąryšių skaičius būtų maksimalus ir kiekvienas žodis turėtų kiek galima daugiau sąryšių (pvz. po 11). Tai įmanoma, nes vienas sąryšis sieja du žodžius.

Blogiausiu atveju fiksavus pirmąjį žodį, visus likusius žodžius galime parinkti 11 būdų. Taip bus, jei sąryšiai sujungs žodžius į grandinę. T. y. pirmasis žodis bus sujungtas vienuolika sąryšių tik su antruoju žodžiu, antrasis – tik su trečiuoju ir t.t. gausime 1120. Visą medį sudarys 1119+1118+...+110 variantų. Paėmę tik didžiausiąjį šios sumos narį 1119 ir nebeskaičiuojant kiek dar gausime variantų antrą, trečią ir t.t. kartą fiksavus pirmąjį žodį, jau galime padaryti tokį įvertinimą:

1119 ( (23)19 ( 257 ( 255. Pasižiūrėję į 159 uždavinyje pateiktą lentelę galime pasakyti, kad algoritmas bus vykdomas ne mažiau 1000 metų... Taigi, šis uždavinys, taip pat priklauso neišsprendžiamų per priimtiną laiką uždavinių klasei. 

168. Atminties paskirstymas. Uždavinio sąlygoje nenurodytas atminties dydžio ribojimas. Susitarsime, kad 1 ( N ( 10. T. y. turimos atminties kiekis (o tuo pačiu ir didžiausias galimas atminties blokų skaičius) neviršys 210 ( 1024ba. 

Informaciją apie kiekvieną bloką patogu saugoti įraše. Jis turės du laukus. Viename jų bus saugomas bloko dydis, kitame – įrašymo operacijos, kurios metu į bloką buvo įrašyti duomenys, numeris. Jei blokas tuščias, įrašymo operacijos numeris lygus nuliui. 

const MAX_BL = 1024; { maksimalus blokų skaičius }
type blokas = record
                bloko_dydis: longint;

                rašymo_op_nr: longint;

              end;

     atmint = array [1..MAX_BL] of blokas;

Sąlygoje nurodyta, kad blokas dalijamas tik į lygias dalis, ir tik vienodo dydžio blokai gali būti sujungiami į vieną. Tai reiškia, kad bet kurio bloko dydis gali būti lygus tik dvejeto laipsniui 2k, k ( 0, 1, 2, ..., 9, 10. 

Todėl daug patogiau operuoti ne su realiu bloko ar duomenų dydžiu, o su atitinkamu dvejeto laipsniu. 

Pavyzdžiui, jei realus bloko dydis lygus 25, tai algoritme jo dydį prilyginame 5. Jei į atmintį reikia įrašyti 9ba, tai mažiausio bloko dydis, į kurį tilps šie duomenys lygus 24. Taigi, įsimename, kad įrašomiems duomenims reikalingas blokas, kurio dydis 4. 

Šis uždavinys nėra įprastinis. Pradiniai duomenys nėra žinomi pradedant vykdyti algoritmą. Jie sužinomi tik vykdymo eigoje. Todėl neįmanoma rasti tokios atminties paskirstymo strategijos, kuri tiktų visiems atvejams. Kiekvienai strategijai galima parinkti tokius pradinius duomenis, kad ta strategija jiems visiškai netiktų.

Pati paprasčiausia strategija būtų tokia. Kai reikia įrašyti duomenis, randam mažiausią laisvą bloką į kurį jie telpa. Jeigu jis per didelis, jį dalijame pusiau kol pasiekiame reikiamą dydį. Įrašome duomenis. Jeigu jis per mažas, sujungiame kuriuos galime tuščius blokus ir tikriname, ar duomenys netilps į kurį nors naujai sujungtą bloką.




168. 3.a–b pav.

Pavyzdžiui, į blokus padalinta atmintis pavaizduota 168.3.a pav. Reikia įrašyti du baitus duomenų. Šie duomenys tilptų į pirmą (jo dydis 2 ba) ir trečią (jo dydis – 4 ba) blokus. Kadangi pirmasis blokas mažesnis už trečią, reiktų rinktis jį. Tačiau jis jau užimtas. Todėl paimame trečią bloką, jį padalijame pusiau ir įrašome duomenis (168.3.b pav.).

Pasirinkome, panašią, tačiau šiek tiek efektyvesnę strategiją. Norėdami įrašyti duomenis ieškosime ne mažiausio tuščio bloko į kurį tilptų duomenys, bet mažiausios tuščių blokų eilės. Imkime tą patį 168.3.a pavyzdį. Yra viena tuščių blokų eilė. Ji užima 1+4+1(6 ba. (168.4.a pav.). Kadangi blokų dydžiai gali būti tik 20, 21, 22, ir t.t., didžiausias blokas, kurį galėsime sudaryti iš 6 ba. užims 22(4 ba. Tačiau mums reikia tik 2 ba. dydžio bloko. Jį ir sudarome (168.4.b–c–d pav.). 




168.a–d pav.

Matome, kad šiuo atveju vietoj dviejų mažesnių „skylių“, turime vieną didesnę.

program atminties_paskirstymas;

  uses atmintis;

const MAX_BL = 1024; { maksimalus blokų skaičius }
type blokas = record
                 bloko_dydis: longint;

                 rašymo_op_nr: longint;

              end;

     atmint = array [1..MAX_BL] of blokas;

{ žemiau aprašyti globalieji kintamieji }
  var N,                         { atminties dydis yra 2^N }
      blokų_kiekis: longint;     { atminties blokų kiekis }
      atm: atmint;               { atmintis }
  procedure sujungti (blk_nr: longint);

    { sujungiami du blokai }
    var i: integer;

  begin
    jungti (blk_nr);

    atm[blk_nr].bloko_dydis := atm[blk_nr].bloko_dydis + 1;

    for i := blk_nr+1 to blokų_kiekis-1 do
      begin
        atm[i].bloko_dydis  := atm[i+1].bloko_dydis;

        atm[i].rašymo_op_nr := atm[i+1].rašymo_op_nr;

      end;

    atm[blokų_kiekis].bloko_dydis  := 0;

    atm[blokų_kiekis].rašymo_op_nr := 0;

    blokų_kiekis := blokų_kiekis - 1;

  end; { sujungti }
  procedure dalinti_pusiau (blk_nr: longint);

    { nurodytasis blokas padalinamas pusiau }
    var i: integer;

  begin
    pusiau (blk_nr);

    atm[blk_nr].bloko_dydis := atm[blk_nr].bloko_dydis - 1;

    { pernumeruojami tolesni blokai }
    for i := blokų_kiekis downto blk_nr + 1 do
      begin
        atm[i+1].bloko_dydis  := atm[i].bloko_dydis;

        atm[i+1].rašymo_op_nr := atm[i].rašymo_op_nr;

      end;

    atm[blk_nr+1].bloko_dydis  := atm[blk_nr].bloko_dydis;

    atm[blk_nr+1].rašymo_op_nr := atm[blk_nr].rašymo_op_nr;

    blokų_kiekis := blokų_kiekis + 1;

  end; { dalinti_pusiau }
  procedure sudaryti_bloką (blk_nr, blk_dydis: longint);

    { sudaro bloką, kurio dydis 2^blk_nr, pradedant nurodytu bloku }
  begin
    if atm[blk_nr].bloko_dydis > blk_dydis { jei blokas per didelis }
       then while atm[blk_nr].bloko_dydis <> blk_dydis do
               dalinti_pusiau (blk_nr)

    else if atm[blk_nr].bloko_dydis < blk_dydis {jei mažesnis nei reikia}
       then begin { sudarome du blokus }
              if (atm[blk_nr].bloko_dydis <> blk_dydis - 1) or
                 (atm[blk_nr+1].bloko_dydis <> blk_dydis - 1)

                 then begin
                        sudaryti_bloką (blk_nr, blk_dydis-1);

                        sudaryti_bloką (blk_nr+1, blk_dydis-1);

                      end;

              sujungti (blk_nr);

            end;

  end; { sudaryti_bloką }
  function dv_laipsn (lp: longint): longint;

    { apskaičiuoja dvejeto laipsnį }
    var rez, i: longint;

  begin
    rez := 1;

    for i := 1 to lp do
      rez := rez * 2;

    dv_laipsn := rez;

  end; { dv_laipsn }
  function log_2 (x: longint): longint;

    var y: longint;

  begin
    y := -1;

    while x > 0 do
      begin
        x := x div 2;

        y := y + 1;

      end;

    log_2 := y;

  end; { log_2 }
  function rasti_laisvą_bloką (b: longint): longint;

    { suranda laisvą bloką, į kurį tilptų b baitų; jei tokio bloko nepavyksta rasti, }

    { funkcijos reikšmė prilyginama nuliui }
    var laips, i, blk_nr, maz, max_dydis, 

        nuo, bl_dydis: longint;

  begin
    { apskaičiuojame reikalingo bloko dydį }
    bl_dydis := 0;

    laips := 1;

    while b > laips do
      begin
        laips := laips * 2;

        bl_dydis := bl_dydis + 1;

      end;

    { randame mažiausią bloką į kurį tilptų mūsų duomenys }
    blk_nr := 0;  maž := N;

    i := 1;

    nuo := 1; { nuo šio bloko prasideda tuščių blokų eilė }
    max_dydis := 0;

    while i <= bloku_kiekis do
      begin
        if atm[i].rašymo_op_nr = 0 { jei tai tuščias blokas }
           then begin

                  max_dydis := max_dydis + 

                               dv_laipsn (atm[i].bloko_dydis);

                end;

        if (atm[i].rašymo_op_nr <> 0) or
           (atm[i].rašymo_op_nr = 0) and (i = blokų_kiekis)

           then begin { jei prasideda užpildytas blokas }
                  if (log_2(max_dydis) >= bl_dydis) and
                     (log_2(max_dydis) <= maž)

                     then begin

                            maž := log_2(max_dydis);

                            blk_nr := nuo;

                          end;

                  max_dydis := 0;

                  nuo := i + 1;

                end;

        i := i + 1

      end;

    if blk_nr <> 0 { jei galime sudaryti reikiamo dydžio bloką }
       then if atm[blk_nr].bloko_dydis > bl_dydis

               { jei turimas blokas per didelis }
               then while atm[blk_nr].bloko_dydis > bl_dydis do
                       dalinti_pusiau (blk_nr)

               else sudaryti_bloką (blk_nr, bl_dydis);

    rasti_laisva_bloką := blk_nr;

  end; { rasti_laisvą_bloką }
  procedure įrašyti (nr, ba: longint; var įvykdyta: boolean);

    { įvykdo įrašymo komandą }
    var bloko_numeris: longint;

  begin
    bloko_numeris := rasti_laisvą_bloką (ba);

    įvykdyta := bloko_numeris <> 0;

    if įvykdyta then begin

                       rašyti (bloko_numeris, ba);

                       atm[bloko_numeris].rašymo_op_nr := nr;

                     end;

  end; { įrašyti }
  var op: char;          { vykdoma operacija }
      nr, b,             { jos nr ir baitų kiekis }
      op_liko: longint;  { tiek operacijų dar reikia įvykdyti }
      įvykdyta: boolean; { ar įvykdyta prieš tai buvusi operacija }
begin
  { sužinome atminties kiekį ir neatliktų operacijų skaičių }
  N := dydis;

  op_liko := skaičius;

  { inicializuojame atmintį }
  blokų_kiekis := 1;

  atm[1].bloko_dydis := N;

  atm[1].rašymo_op_nr := 0;

  įvykdyta := true;

  while įvykdyta and (op_liko > 0) do
    begin
      oper (op, nr, b);  { sužinoma tolesnė operacija }
      if op = 'T'

         then begin { ištrinami duomenys }
                atm[nr].rašymo_op_nr := 0;

                trinti (nr);

              end
         else įrašyti (nr, b, įvykdyta);

      op_liko := op_liko - 1;

    end;

  baigti;

end.

169. Šimtalangės šaškės. Tai perrinkimo uždavinys. Neįpratusiems rašyti perrinkimo algoritmus kils klausimas. Kaipgi organizuoti perrinkimą, jei per vieną rekursijos žingsnį. šaškė gali padaryti labai daug skirtingų ėjimų: juk ji gali bet kada nustoti kirsti. 

Tam įvedama nauja sąvoka žingsnis. Per vieną žingsnį šaškė paeina iš langelio kuriame stovi į gretimą (jei nekerta) arba peršoka gretimą langelį (jei kerta). Jei šaškė nekerta arba kerta tik vieną šaškę, tuomet vieną ėjimą sudaro vienas žingsnis, jei kerta dvi ar daugiau šaškių – keli žingsniai. Dabar per vieną rekursijos žingsnį šaškė padarys vieną žingsnį. T. y. jei yra galimybe kirs atgal į kairę ir atgal į dešinę bei paeis (arba kirs) pirmyn į kairę ir pirmyn į dešinę. Įsimenant kiekvieną padarytą žingsnį reikės kartu įsiminti ir kuriam ėjimui priklauso tas žingsnis.

Atliekant naują žingsnį reikės nustatyti ar tai tas pats ėjimas, ar jau naujas. Tam rekursinėje procedūroje įvedama naujas parametras–reikšmė, kuris lygus nuliui, jei praeitu žingsniu nebuvo kirsta arba baltosios šaškės numeriui, kuria buvo kirsta praeitu žingsniu. 

Jei praeitu žingsniu nebuvo kirsta (169.5.a pav.) arba šiuo žingsniu nekertama (169.5.b pav.), tai tikrai atliekamas naujas ėjimas. Jei ir praeitu ir šiuo žingsniu kirsta, tai naujas ėjimas bus tik tada jei sutaps kertančiosios baltos šaškės numeriai (165.c–d. pav.). 




169.5.a pav.




169.5.b. pav.




169.5.c pav. Tęsiamas tas pats ėjimas




169.5.d pav. Naujas ėjimas

Belieka įvertinti vykdymo laiko eilę. Apie šią sąvoką galite pasiskaityti prie 159 uždavinio sprendimo. Vienu rekursijos žingsniu šaškė gali padaryti daugiausia keturis skirtingus žingsnius. Lentoje gali būti iki trijų baltų šaškių. Jau gauname 12 žingsnių. Viso sprendinį gali sudaryti iki 15–os ėjimų. Blogiausiu atveju sprendinį gali sudaryti 14+20(34 žingsniai. Taip bus jei bus nekertant paeita 14 kartų ir penkioliktu ėjimu nukirsta 20 juodųjų šaškių. Taigi, blogiausiu atveju medžio viršūnėje bus 1234 rekursinių žingsnių. Be abejo, tai gan grubus įvertinimas. Juk ne kiekvienu atveju bus įvykdomi visi 34 rekursijos žingsniai. 

Iš 159 uždavinyje pateiktos lentelės matome, kad algoritmas, kurio eilė 255 būtų vykdomas apie 1000 metų. Na o 1234 ( (23)34 ( 2102.....

Belieka pridurti, kad šio uždavinio per priimtiną laiką išspręsti neįmanoma. O aukščiau aprašytasis sprendimas greitai pabaigia darbą tik su nedideliais arba specifiniais testais.

const MAX = 200;  { maksimalus žingsnių skaičius }
      DYDIS = 50; { juodų langelių skaičius lentoje }
type langelis = (balta, juoda, nieko, už_ribų);

     lenta = array [0..DYDIS] of langelis; 

                                                { nulinis lentos langelis visuomet yra už ribų }
     lentelė = array [1..MAX] of integer;

     h_kryptis = (kairėn, dešinėn);

     v_kryptis = (žemyn, aukštyn);

{ žemiau aprašyti globalieji kintamieji }
var b, j: integer; { baltų bei juodų šaškių skaičius lentoje }
    bal: lentelė;  { baltų šaškių vieta lentoje }
    l: lenta;      { šaškių išsidėstymas lentoje }
    žin,           { padarytų žingsnių skaičius }
    ėj,            { padarytų ėjimų skaičius }
    min_žin, min_ėj : integer; { minimalūs ėjimų bei žingsnių skaičiai }
    žings,         { padarytieji žingsniai }
    ėj_nr,         { žingsnius atitinkančių ėjimų numeriai }
    min_ėj_nr, min_žings: lentelė;  { padarytieji ėjimai }
function koks_langelis (lang: integer; 

                        h: h_kryptis; v: v_kryptis): integer;

  { nustato į kokį langelį pakliūsime, jei iš langelio lang paeisime nurodyta kryptimi }
  var liek: integer;

begin
  liek := lang mod 10;

  if (h = kairėn) and (liek = 1) or (h = dešinėn) and (liek = 0)

    then lang := 0 { už lentos ribų }
    else begin

           if v = aukštyn

              then lang := lang + 5

              else lang := lang - 5;

           if (h = kairėn) and (liek >= 1) and (liek <= 5)

              then lang := lang - 1

           else if (h = dešinėn) and ((liek = 0) or (liek > 5))

              then lang := lang + 1;

         end;

  if (lang >= 1) and (lang <= 50)

     then koks_langelis := lang

     else koks_langelis := 0 { už lentos ribų }
end; { koks_langelis }
procedure tapti_dama;

  { pagrindinė procedūra }
  { jos pradiniai duomenys: b, j, bal, l }
  { jos gauti rezultatai – min_žin, min_ėj, min_ėj_nr, min_žings }
  procedure eiti (ėj, žin, kirsta: integer 

                           { kuria baltąja šaške buvo kirsta praeitu žingsniu});

    { rekusrinė procedūra, atliekanti visu įmanomus ėjimus }
    var dama: boolean; { ar baltoji šaškė jau yra dama }
        i, lan1, lan2, buvęs: integer;

        horiz: h_kryptis;

        vert: v_kryptis;

  begin
    { patikrinsime, gal nors viena baltoji šaškė jau yra dama }
    dama := false;

    for i := 1 to b do
      dama := dama or (bal[i] >= 46);

    { jei baltoji šaškė yra dama, gal radome trumpesnį kelią }
    if dama and (ėj < min_ėj)

       then begin { įsimename gautą rezultatą }
              min_žings := žings;

              min_žin := žin;

              min_ėj_nr := ėj_nr;

              min_ėj := ėj;

            end;

    { jei šaškė tapo dama arba viršytas minimalus ėjimų skaičius, rekursija sustabdoma }
    if dama or (ėj >= min_ėj) then exit;

    { jei nė viena baltoji šaškė dar nėra dama }
    for i := 1 to b do { eisime kiekviena balta šaške }
     for horiz := kairėn to dešinėn do { bandysime eiti visomis keturiomis }
       for vert := žemyn to aukštyn do { kryptimis }
         begin
           { randame du gretimus langelius }
           lan1 := koks_langelis (bal[i], horiz, vert);

           lan2 := koks_langelis (lan1, horiz, vert);

           { bandome paeiti arba kirsti }
           if (l[lan1] = nieko) and (vert = aukštyn)

              then begin { bandome paeiti: naujas ėjimas }
                     žings[žin+1] := bal[i];

                     ėj_nr[žin+1] := ėj + 1;

                     žings[žin+2] := lan1;

                     ėj_nr[žin+2] := ėj + 1;

                     buvęs := bal[i];

                     bal[i] := lan1;

                     l[lan1] := balta;

                     l[buvęs] := nieko;

                     eiti (ej+1, žin+2, 0);

                     { atstatome reikšmes }
                     l[lan1] := nieko;

                     l[buvęs] := balta;

                     bal[i] := buvęs;

                   end
              else if (l[lan1] = juoda) and (l[lan2] = nieko)

                then begin { bandome kirsti }
                       buvęs := bal[i];

                       bal[i] := lan2;

                       l[buvęs] := nieko;

                       l[lan1] := nieko;

                       l[lan2] := balta;

                       if kirsta = i { jei tęsiasi tas pats ėjimas }
                          then begin
                                 žings[žin+1] := lan2;

                                 ėj_nr[žin+1] := ėj;

                                 eiti (ėj, žin+1, i)

                               end
                          else begin { jei tai naujas ėjimas }
                                 žings[žin+1] := buvęs;

                                 ėj_nr[žin+1] := ėj+1;

                                 žings[žin+2] := lan2;

                                 ėj_nr[žin+2] := ėj + 1;

                                 eiti (ėj+1, žin+2, i);

                               end;

                       { atstatome reikšmes }
                       bal[i] := buvęs;

                       l[buvęs] := balta;

                       l[lan1] := juoda;

                       l[lan2] := nieko;

                     end;

           end;

  end; { eiti }
  var užimta: boolean;

      i: integer;

begin
  { jei paskutinioji eilė užimta juodomis šaškėmis – sprendinio nėra }
  užimta := true;

  for i := 46 to 50 do
    užimta := užimta and (l[i] = juoda);

  min_ėj := 16;    { sprendinyje ėjimų skaičius neviršys 15 }
  ėj := 0;         { dar nepadarytas nė vienas ėjimas }
  žin := 0;        { ir nė vienas žingsnis }
  if not užimta then eiti (ej, 0, 0); { rekursiškai ieškome sprendinio }
  if min_ėj = 16   { jei sprendinio neradome }
     then min_ėj := 0;

end; { tapti_dama }
Žemiau esančiuose paveikslėliuose pateikiami šio uždavinio testai. 

	Tes​to nr.
	Pradiniai duomenys
	Galimi rezultatai
	Paaiškinimai

	0
	3

1 3 4

14

2 6 8 9 20 21 30 31 40 46 47 48 49 50
	0


	Sprendinių nėra.

	1
	2

25 27

14

6 7 11 12 17 23 30 31 32 39 40 42 45 47
	3

27 36

36 41

41 46
	Paprastas pavyzdys, vienas kirtimas.

	2
	2

1 2

14

10 11 14 18 19 23 24 28 31 41 46 47 48 50
	9

1 6

6 12

12 16

16 22

22 27

27 33

33 38

38 44

44 49
	Sprendimas nekertant.

	3
	3

12 16 22

14

14 19 23 26 31 33 37 38 39 40 42 45 47 48
	4

12 17

16 21

21 32 43

17 28 37 46
	Reikia panaudoti dvi šaškes.

	4
	2

9 36

15

12 13 14 15 22 23 25 31 32 34 35 43 45 49 50
	1

9 20 29 38 47
	Daug kirtimų: reikia pasirinkti teisingą kirtimų seką.

	5
	3

7 10 23

16

6 13 24 29 31 34 36 37 38 39 44 45 46 47 48 50
	6

23 28

28 19

19 24

24 35

35 40

40 49
	Daug šaškių; reikia kirsti atgal.

	6
	1

4

15

7 9 12 18 19 22 28 29 32 38 39 42 46 48 49
	3

4 15 24 13 2

2 7

7 16 27 36 47
	Dvi ilgos kirtimų sekos.

	7
	3

4 5 23

20

9 12 13 14 17 24 26 27 28 31 32 34 37 38 42 44 47 48 49 50
	9

4 15

5 9

9 18 7 16

16 22

22 33

23 27

27 36

36 41

41 46
	Reikia panaudoti tris šaškes.

	8
	3

4 41 42

20

7 8 9 13 15 16 17 23 25 26 27 33 34 36 37 43 46 47 49 50
	5

41 32 21 12 3 14 5
4 9

9 20 29 38

38 44

44 48
	Sudėtingas atvejis: ėjimų kryptys neprognozuo​jamos.


Testai iliustruojami žemiau pateiktuose paveikslėliuose.

0 testas.



1 testas



2 testas



3 testas. 



4 testas



5 testas



6 testas



7 testas



8 testas



170. Pakeisk žodį. Antrojo žodžio ilgį pažymėkime n. Parodysime, kad norint iš vieno žodžio gauti kitą reikia ne mažiau n operacijų.

Operacijų skaičius bus mažiausias jei abu žodžiai sutaps. Tuomet tereiks n kartų perkelti žymeklį kol jis atsidurs ant žodžio pabaigos simbolio. 

Jei žodžiai nesutampa į pirmąjį žodį galime nekreipti dėmesio ir prieš pirmąją jo raidę įterpti visą antrąjį žodį n. Tai sudarys n operacijų. Numetę nereikalingą žodžio dalį gausime reikiamą žodį. Tai sudarys n+1 operaciją. 

	batas -> batas
	batas
	
	molis -> batas
	molis

	Perkelti
	batas
	
	Įterpti(b)
	bmolis

	Perkelti
	batas
	
	Įterpti(a)
	bamolis

	Perkelti
	batas
	
	Įterpti(t)
	batmolis

	Perkelti
	batas
	
	Įterpti(a)
	batamolis

	Perkelti
	batas_
	
	Įterpti(s)
	batasmolis

	
	
	
	Numesti
	batas_


Gauname, kad minimalus operacijų skaičius visada bus lygus arba n arba n+1. Operacijų skaičius bus lygus n tik tada, jei prieš atliekant i-tąją operaciją keičiamo žodžio pradžią sudarys pirmųjų i antrojo žodžio raidžių, o žymeklis rodys į i-tąją raidę. 

Panagrinėkime duotąsias operacijas. Tarkime, iš žodžio balkonasxxx mums reikia gauti žodį baritonas. Šiuo atveju n = 9. 
	Kuri operacija bus atliekama
	Žodis prieš atliekant operaciją ir ją atlikus
	Kelinta operacija bus atliekama 
	Kelintą žodžio raidę rodo žymeklis prieš atliekant tą operaciją.
	Kelintą žodžio raidę rodo žymeklis atlikus tą operaciją.

	Perkelti
	balkonasxxx –> balkonasxxx
	1
	1
	2

	Perkelti
	balkonasxxx –> balkonasxxx
	2
	2
	3

	Pašalinti
	balkonasxxx –> bakonasxxx
	3
	3
	3

	Pakeisti(r)
	bakonasxxx –> baronasxxx
	4
	3
	3

	Perkelti
	baronasxxx –> baronasxxx
	5
	3
	4

	Įterpti(i)
	baronasxxx –> barionasxxx
	6
	4
	5

	Įterpti(t)
	barionasxxx –> baritonasxxx
	7
	5
	6

	Perkelti
	baritonasxxx –> baritonasxxx
	8
	6
	7

	Perkelti
	baritonasxxx –> baritonasxxx
	9
	7
	8

	Perkelti
	baritonasxxx –> baritonasxxx
	10
	8
	9

	Perkelti
	baritonasxxx –> baritonasxxx
	11
	9
	10

	Numesti
	baritonasxxx –> baritonas_
	12
	10
	Žodžio pabaigos simbolis (10)


Matome, kad atliekant operacijas 

Pastudijavę duotas operacijas pamatysime, kad atlikinėti operacijų Pašalinti, Pakeisti(x), Numesti neverta: jei prieš atliekant šias operacijas žymeklis rodė i-tąjį simbolį, tai ir atlikus šias operacijas jis taip pat rodys i-tąjį simbolį.

Vadinasi, n operacijų užteks tik tada, jei iš pirmojo žodžio galėsime gauti antrąjį taikydami tik operacijas Perkelti arba Įterpti(x). 

Atkreipkite dėmesį, kad jei nesutampa žodžių paskutiniosios raidės, šių operacijų neužteks. Mat nebegalima įterpti stovint ant žodžio pabaigos simbolio. Pavyzdžiui, iš žodžio ji norime gauti jis.

	Operacija
	Žodis

	
	ji

	Perkelti
	ji

	Įterpti(i)
	jii

	Įterpti(s)
	jisi

	Numesti
	jis_


function keisti (žod1, žod2: string): integer;

  var kiek, i: integer;

begin
  kiek := length (žod2);

  if (length (žod1) = length (žod2)) and (žod1 <> žod2) 

    or (length (žod1) > length (žod2)) 

    or (žod1[length(žod1)] <> žod2[length(žod2)])

    then kiek := kiek + 1

    else 

      if length (žod1) < length (žod2) { pirmas žodis trumpesnis }
       then begin
             žod1 := žod1 + '.'; { abu žodžiai baigsis tuo }
             žod2 := žod2 + '.'; { pačiu simboliu }
             i := 1;

             while i <= length (žod2) do
               begin
                 if žod1[i] <> žod2[i]

                  then insert (žod2[i], žod1, i); { įterpiame }
                 i := i + 1;                     { perkeliame }
               end;

             if žod1 <> žod2

                then kiek := kiek + 1

            end;

  keisti := kiek

end; { keisti }
171. Geležinkelio grotelės. Uždavinio sprendimo idėją paaiškinsime remdamiesi sąlygoje pateiktu pavyzdžiu. Duotasis teksto fragmentas gali prasidėti bet kuriame stulpelyje nuo pirmojo iki paskutiniojo.

Pradžioje laikome, kad tekstas prasideda fragmentu. Nežinomą tekstą bei fragmentą įrašome į groteles: 

	S
	
	
	
	
	
	
	
	
	
	?
	
	
	
	

	
	K
	
	
	
	
	
	
	
	?
	
	?
	
	
	

	
	
	A
	
	
	
	
	
	?
	
	
	
	?
	
	

	
	
	
	M
	
	
	
	?
	
	
	
	
	
	?
	

	
	
	
	
	B
	
	?
	
	
	
	
	
	
	
	?

	
	
	
	
	
	?
	
	
	
	
	
	
	
	
	


171.4 pav.

ŠBN

PAI

AI

_MS

S

AKU

171.5 pav. Duoto​sios teksto eilutės
S?

K??

A??

M??

B??

?

171.6 pav. Už​ko​du​otas frag​mentas

Grotelėse klaustukais pažymėjome likusius simbolius, kurie būtų tekste, jei frag​mentas prasidėtų pir​ma​ja​me stulpelyje. Už​šif​ruo​jame šį „tekstą“ (171.6 pav.). Kiekvienai šio naujo teksto eilutei ieškome ati​tik​mens tarp turimų eilučių. 

Pirmajai eilutei nepavyksta rasti atitikmens: tarp duotųjų nėra nė vienos eilutės, kurią sudarytų du simboliai ir kurių pirmasis būtų raidė S.

Tuomet tekstą šifruojame laikydami kad fragmentas prasideda antrajame stulpelyje:

	?
	
	
	
	

	
	S
	
	
	
	
	
	
	
	?
	
	?
	
	
	

	
	
	K
	
	
	
	
	
	?
	
	
	
	?

	M

	B
	
	
	
	
	
	
	
	
	


171.7 pav.

??

S??

K??

A??

M??

B

171.8 pav.

Gauname tokias eilutes:

Pirmąją naujai užkoduoto teksto eilutę atitiktų eilutė AI (t. y. bet kokia eilutė iš dviejų simbolių). Deja, antrajai eilutei atitikmens nerandame.

??

?B?

?M?

?A?

?K?

S

171.9 pav.

Ir taip fragmentą stumiame tekstu kol randame tinkamą sprendinį. 

Mūsų pavyzdyje rasime sprendinį, kai fragmentas prasidės šeštame stulpelyje ir turėsime tokias naujai užšifruotas eilutes:

?S

??K

??A

??M

??B

?

171.10 pav. Už​ko​du​otas frag​mentas

Pats tekstas gali būti gana ilgas: iki 12(255 ( 3060 simbolių. Blogiausiu atveju gali tekti užšifruoti tiek kartų. Tai užima nemažai laiko. Algoritmas pagreitės, jei vieną kartą užšifravę nežinomą tekstą su fragmentu, iš karto išanalizuosime visus atvejus, kai fragmentas prasideda toje pačioje eilutėje ir eina ta pačia kryptimi (t. y. arba leidžiasi, arba kyla). Imkime 171.6 pav. Jei nuo pirmosios ir paskutiniosios eilutės galo nutrinsime po vieną nežinomą simbolį (klaustuką), o nuo likusių eilučių – po du ir kiekvienos eilutės pradžioje prijungsime tiek nežinomų simbolių (klaustukų) kiek jų buvo nutrinta – gausime užšifruotą tekstą su fragmentu, kai fragmentas prasideda pirmoje eilutėje, bet nebe pirmame , o dešimtame stulpelyje.

Šitaip patobulinus algoritmą užšifruoti nežinomą tekstą su fragmentu reikės ne daugiau 2(k – 3 kartų. T. y. blogiausiu atveju ne daugiau 2(12–3(21 karto.

const MAX_K = 12; { maksimalus aukštis }
      MAX_ILG = 255; { maksimalus eilutės ilgis }
type  eilutės = array [1..MAX_K] of string;

      lentelė = array [1..MAX_K] of integer;

      aibė = set of 1..MAX_K;

      aib_mas = array [1..MAX_K] of aibė;

      log_lentelė = array [1..MAX_K] of boolean;

{ žemiau aprašyti globalieji kintamieji }
var k: integer;  { aukštis }
    fr: string;  { fragmentas }
procedure šifruoti (teksto_il, nr: integer; 

                    var fr_eil: eilutės);

  { užšifruojamas fragmentas. Pirmoji užšifruota raidė dedama į stulpelį su numeriu nr. }
  var i, eilutė: integer;

      žemyn: boolean;

begin
  for i := 1 to k do
    fr_eil[i] := '';

  žemyn := true;

  eilutė := 1;

  for i := 1 to teksto_il do
    begin
      if (i < nr) or (i > nr + length(fr) - 1)

         then fr_eil[eilutė] := fr_eil[eilutė] + chr(1)

         else fr_eil[eilutė] := fr_eil[eilutė] + 

                                                fr[i-nr+1];

      if (eilutė = 1) and (i <> 1) or (eilutė = k)

         then žemyn := not žemyn;

      if žemyn

         then eilutė := eilutė + 1

         else eilutė := eilutė - 1;

    end;

end; { šifruoti }
procedure stumti (var fr_eil: eilutės; 

                  var pavyko: boolean);

  var i, j: integer;

begin
  pavyko := true;

  for i := 1 to k do
    for j := 1 to 2 do
      if (j <> 2) or (i <> 1) and (i <> k)

         then begin
                if fr_eil[i, length(fr_eil[i])] <> chr(1)

                   then pavyko := false;

                delete (fr_eil[i], length(fr_eil[i]), 1);

                fr_eil[i] := chr(1) + fr_eil[i];

              end;

end; { stumti }
procedure tinka (post: integer; fr_eil: eilutės;

                 var rasta: boolean; var eil: eilutės);

  var tiks: aib_mas;

      buvo: log_lentelė;

  procedure dėlioti (j: integer; 

                     var rasta: boolean; var rez: eilutės);

    { rekursinė procedūra, ieškanti atitikimo tarp užšifruoto fragmento ir teksto ei​lu​čių }
    var i: integer;

  begin
    for i := 1 to k do
      if not buvo[i] and (i in tiks[j]) and  not rasta

         then begin
                buvo[i] := true;

                rez[j] := eil[i];

                if j = k

                   then rasta := true

                   else dėlioti (j+1, rasta, rez);

                { atstatome buvusias reikšmes }
                if not rasta

                   then buvo[i] := false;

              end;

  end; { dėlioti }
  var i, j, m, mm: integer;

      rasta_fr, rasta_eil: aibė;

      rez: eilutės;

begin
  { sužinome kurios teksto eilutės kuriomis užšifruoto fragmento eilutėmis gali būti }
  for j := 1 to k do
    tiks[j] := [];

  for j := 1 to k do
   begin
     if (j = 1) or (j = k) 

        then m := 1

        else m := post;

     while (fr_eil[j,m] = chr(1)) and 

           (m < length(fr_eil [j])) do
       m := m + 1;

     for i := 1 to k do
       if length(eil[i]) = length(fr_eil[j])

          then begin  { ar i–oji teksto eilutė gali būti j–ają fragmento eilute }
                 mm := m;

                 while (fr_eil[j, mm] = eil[i, mm]) and 

                       (mm < length(eil[i])) do
                   mm := mm + 1;

                 if (fr_eil[j, mm] = chr(1)) or 

                    (fr_eil[j, mm] = eil[i, mm])

                     then tiks[j] := tiks[j] + [i];

               end;

     { jei kažkuriai eilutei neradome atitikmens – nutraukiame darbą }
     if tiks[j] = [] then break;

   end;

  { patikriname, ar kiekvienai eilutei rastas atitikmuo }
  rasta_fr := []; rasta_eil := [];

  for i := 1 to k do
    begin
      if tiks[i] <> [] then rasta_fr := rasta_fr + [i];

      rasta_eil := rasta_eil + tiks[i];

    end;

  { rekursiškai pabandome rasti atitikimą }
  for i := 1 to k do  { pasiruošimas rekursijai }
    buvo[i] := false;

  if (rasta_fr = [1..k]) and (rasta_eil = [1..k])

     then begin
            dėlioti (1, rasta, rez);

            if rasta then eil := rez;

          end
     else rasta := false;

end; { tinka }
procedure ieškoti (var eil: eilutės);

  { pagrindinė uždavinį sprendžianti procedūra }
  var post, stu, i, teksto_il: integer;

      rasta, pavyko: boolean;

      fr_eil: eilutės;

begin
  { suskaičiuojame teksto ilgį }
  teksto_il := 0;

  for i := 1 to k do
    teksto_il := teksto_il + length(eil[i]);

  rasta := false;

  stu := 1;

  šifruoti (teksto_il, stu, fr_eil);

  post := 1;

  while not rasta and (stu <= 2*k-2) do
    begin
      { patikriname, ar fragmentas tinka }
      tinka (post, fr_eil, rasta, eil);

      { užšifruojame fragmentą }
      stumti (fr_eil, pavyko);

      if not pavyko

         then  begin
                 stu := stu + 1;

                 post := (stu-1) div k + 1;

                 šifruoti (teksto_il, stu, fr_eil);

               end
         else post := post + 2;

    end;

end; { ieškoti }
172. Svečiai iš Japonijos. Šį uždavinį galime nesunkiai suvesti į Maksimalių srautų uždavinį, kuris buvo aprašytas prie 111 uždavinio sprendimo. 

Kiekvienas moksleivis atitiks mazgą grafe. Jei moksleivis A sutinka priimti (gyventi pas) moksleivį B, tai tuos mazgus jungia briauna. Akivaizdu, kad mazgai atitinkantys du japonus (ar du lietuvius) negali būti sujungti briauna. Imkime sąlygoje duotą pavyzdį. Gausime tokį grafą:




172.1 pav.

Įveskime papildomus pradinį bei galinį mazgus. Pradinį mazgą sujunkime su visais mazgais, atitinkančiais lietuvių moksleivius, galinį – su mazgais atitinkančiais japonų moksleivius. Visų briaunų svorius prilyginkime 1. Gauname:




172.2 pav.

Kiekvienas srautas ištekėjęs iš pradinio mazgo ir pasiekęs galinį reiškia patenkintą dviejų moksleivių pageidavimą gyventi kartu. Taigi, radę maksimalų srautą šiame grafe kartu rasime ir maksimalų patenkintų moksleivių skaičių. Šio algoritmo eilė (žr. 159 užd.) lygi (n+m)3.

Tačiau šis grafas yra ypatingas. Tai dvidalis grafas, kurio visų briaunų svoriai lygūs 1. Maksimalaus srauto tokiame grafe paieškai galime taikyti efektyvesnį Hopkrofto–Karpo algoritmą, kurio eilė lygi O((n+m+br)(br). Čia br – briaunų skaičius minėtame grafe.

Algoritmo esmė tokia. Imamas pradinis grafas ir sudaromas bet koks sprendinys:




172.3 pav.

Visos grafo briaunos paverčiamos lankais. Sprendiniui priklausančių briaunų kryptis yra iš dešinės į kairę, sprendiniui nepriklausančių – iš kairės į dešinę. Prijungiama pradinė viršūnė. Iš jos nuvedami lankai į kairėje esančias ir į sprendinį neįtrauktas viršūnes, o iš dešinėje esančių ir į sprendinį neįtrauktų viršūnių nuvedami lankai į galinę viršūnę.




172.4 pav.

Ieškomas trumpiausias kelias iš pradinės viršūnės į galinę. Šis kelias būtinai turės eiti per vieną laisvą kairės ir per vieną laisvą dešinės pusės viršūnę. Kas antras tokio kelio lankas (nekaitant lanko vedančio iš pradinės viršūnės ir lanko, vedančio į galinę viršūnę) priklausys iki šiol rastam sprendiniui:




172.5 pav.

Turime tokį kelią: P–>5K–>3D–>3K–>2D–>G (raidės K ir D nurodo kairėje ar dešinėje grafo dalyje yra nagrinėjamos viršūnės). Imame visus šio kelio lankus (išskyrus lankus, jungiančius su pradine ir galine viršūne) ir keičiame jų paskirtį: jei lankas priklausė sprendiniui, jis pašalinamas iš sprendinio, jei nepriklausė sprendiniui – įtraukiamas į jį:




172.6 pav.

Po vieno tokio žingsnio (trumpiausio kelio suradimo ir sprendinio atnaujinimo) į sprendinį įtraukiamos dar dvi viršūnės. Veiksmai kartojami tol, kol pavyksta rasti trumpiausią kelią iš pradinės viršūnė į galinę. Jei tokio kelio nebepavyktų rasti (pvz., jo nebeapvyks rasti 172.6 pav. atveju), tai reiškia, kad rastasis sprendinys yra optimalus. 

Algoritme nebus kuriamos papildomos pradinė ir galinė viršūnės, o trumpiausio kelio bus ieškoma nuo laisvų kairės pusės viršūnių iki bet kurios laisvos dešinės pusės viršūnės. Paties kelio bus ieškoma taikant paiešką į plotį. Rezultatui (t.y. porų sąrašas) saugoti bus sukurtos dvi viena kitai simetriškos vienmatės lentelės K ir D. Jei K[u] ( v (tuomet būtinai D[v] ( u), tai reiškia, kad sprendiniui priklauso pora (u, v), kur viršūnė u yra iš kairės pusės, o viršūnė v – iš dešinės pusės. Pačiam rezultatui saugoti užtektų vienos iš jų, tačiau darbo metu patogiau turėti jas abi. 

Rastąjį trumpiausią kelią nusakys dvi lentelės: tai lentelė kel_K bei jau minėta lentelė D. Imkime kelią, pavaizduotą 172.5 pav. Lentelės atrodys taip:

	nr
	1
	2
	3
	4
	5

	kel_K
	0
	0
	5
	0
	0


	nr
	1
	2
	3
	4

	D
	2
	0
	3
	0


Žinome, kad kelias baigiasi trečia viršūnė kairėje ir trečia viršūne dešinėje. Pasižiūrėję į kel_K, sužinome, kad į 3K (t.y. trečią viršūnę kairėje) atėjome iš 5K. Pasižiūrėję į lentelės D elementą D[3], sužinome, kad buvo eita per viršūnę 3D. Taip gavome kelio atkarpą: 5K–>3D–>3K. Taip po žingsnį eidami atgal sužinome visą kelią. 

Reikia nepamiršti, kad privaloma apgyvendinti visus svečius. Taigi, kai daugiau pageidavimų patenkinti neįmanoma, likę japonų moksleiviai apgyvendinami pas bet kuriuos dar neturinčius svečių vietinius moksleivius.

program svečiai_iš_Japonijos;

  const MAX_M = 50;   { max(|K|, |D|) }
  type lentelė = array [0..2*MAX_M] of integer;

       grafas = array [1..MAX_M, 1..MAX_M] of boolean;

  { žemiau aprašyti globalieji kintamieji }
  var g: grafas; { pradinis grafas }
      n, m: integer; { lietuvių bei japonų moksleivių skaičius }
      K, D: lentelė; { ieškomasis sprendinys; D[v] rodys numerį moksleivio, kuris }

                     { apgyvendins svečią v }
  {--------------- Procedūros darbui su eile ------------------------}
  procedure įtraukti_į_eilę (u: integer;  var Q:lentelė;

                             var uodega: integer);

    { įtraukia į eilės pabaigą viršūnę u }
  begin
    Q[uodega] := u;

    uodega := uodega + 1;

  end; { įtraukti_į_eilę }
  procedure pašalinti_iš_eilės (var galva: integer);

  begin
     galva := galva + 1

  end; { pašalinti_iš_eilės }
  {--------------- Rezultato suformavimas --------------------------------}

  procedure papildyti (liet: lentelė; var japo: lentelė);

   { kai daugiau pageidavimų patenkinti neįmanoma, likę moksleiviai
    apgyvendinami neatsižvelgiant į jų norus }
    var u, v: integer;

  begin
    u := 1;

    for v := 1 to m do
      if japo[v] = 0   { jei svečias dar neapgyvendintas }
         then begin
                while liet[u] <> 0 do { ieškome svečio neturinčio lietuvio }
                   u := u + 1;

                 japo[v] := u; liet[u] := v;  { apgyvendiname }
              end;

  end; { papildyti }
  {=============== Maksimalaus srauto paieškos procedūros ==========}
  procedure sprendinio_sudarymas (var K, D: lentelė);

    { sudaro pirmą pasitaikiusį sprendinį }
    var u, v: integer;

  begin
    { inicializuojame lenteles K ir D}
    for u := 1 to n do
      K[u] := 0;

    for v := 1 to m do
      D[v] := 0;

    { sudarome sprendinį }
    for u := 1 to n do
      begin
        v := 1;

        while (v <= m) and (K[u] = 0) do
          begin { kol nerasta pora arba neišnagrinėti visi variantai }
            if g[u, v] and (D[v] = 0)

               then begin
                      K[u] := v; D[v] := u;

                    end;

            v := v + 1;

          end;

      end;

  end; { sprendinio_sudarymas }
  {--------------------------- Naujo kelio paieška-----------------------}
  procedure inicializuoti (var kel_K, ilgis: lentelė;

                           var Q: lentelė; 

                           var galva, uodega: integer);

   { inicializuojame kel_k ir eilę Q (į Q įtraukiamos laisvos viršūnės iš K)}
    var u: integer;

  begin
    { inicializuojamas kelias }
    for u := 1 to n do
      begin
        kel_K[u] := 0;

        ilgis[u] := maxint;

      end;

    { inicalizuojama eilė, į ją įtraukiamos laisvos viršūnės iš K}
    galva := 1; uodega := 1;

    for u := 1 to n do
      if K[u] = 0

         then begin
                įtraukti_į_eilę (u, Q, uodega);

                ilgis[u] := 0;

              end;

  end; { inicializuoti }
  procedure ieškoti_kelio (var kel_K: lentelė;   { rastasis kelias }
                          var pask_K, pask_D: integer;

                          { rastojo kelio pabaigos viršūnės }
                          var rasta: boolean);{ ar pavyko surasti kelią }
    { ieškomas kelias nuo bet kurios laisvos viršūnės iš K iki bet kurios
     laisvos viršūnės iš D }
    var ilgis,  { ilgis[u] - trumpiausio kelio iki u ilgis }
        Q: lentele; { eilė }
        galva, uodega: integer; { eilės pradžia ir pabaiga }
        u, v: integer;

  begin
    { inicializuojame kel_k ir eilę Q (į Q įtraukiamos laisvos viršūnės iš K)}
    inicializuoti (kel_k, ilgis, Q, galva, uodega);

    { vykdoma paieška į plotį }
    rasta := false; { kelias dar nerastas }
    while (galva <> uodega) and not rasta do { kol eilė netuščia }
      begin                                  { ir nerastas kelias }
        u := Q[galva]; { imamas pirmas eilės elementas }
        v := 1;

        while (v <= m) and not rasta do { ieškome lankų iš u į K viršūnes }
          begin
            if g[u, v] and 

              ((D[v] = 0) or (ilgis[u]+2 < ilgis[D[v]]))

              { jei veda lankas iš u į v ir kelias trumpesnis nei
               iki šiol rastasis arba prieita kelio pabaiga }
              then begin  { u(v(u' }
                    kel_K[D[v]] := u;        { iš u eisime į u'per v }
                    ilgis[D[v]] := ilgis[u]+2; { atnaujiname ilgį iki u' }
                    { u'  bus įtrauktas į eilę }
                    įtraukti_į_eilę(D[v], Q, uodega); 

                    rasta := D[v] = 0;

                  end;

             v := v + 1

           end; { while (v ( m) and not rasta }
        pašalinti_iš_eilės (galva);

      end;

    pask_K := u; pask_D := v-1;

  end; { ieškoti_kelio }
  {--------------------------- Naujo kelio paieškos pabaiga ---------------}
  procedure atnaujinti_sprendinį (var K, D: lentelė; 

                                  kel_K: lentelė;

                                  u, v: integer);

    { suradus naują kelią, kurį nusako lentelės kel_K ir D,
     atnaujinamas sprendinys }
  begin
    repeat
      K[u] := v; D[v] := u;

      u := kel_K[u];

      v := K[u];

    until u = 0;

  end; { atnaujinti_sprendinį }
  procedure max_srautas (var K, D: lentelė);

    { pagrindinė procedūra, ieškanti maksimalaus srauto }
    var kel_K: lentelė;    { kur ves naujas kelias iš aibės K viršūnių }
        pask_K, pask_D: integer;  { dvi paskutinės kelio viršūnės }
        rasta: boolean;    { ar rastas naujas kelias }
  begin
    { sudarome bet kokį sprendinį }
    sprendinio_sudarymas (K, D);

    ieškoti_kelio (kel_K, pask_K, pask_D, rasta);

    while rasta do { kol pavyksta rasti kelią }
      begin
        atnaujinti_sprendinį (K, D, kel_K, pask_K, pask_D);

        ieškoti_kelio (kel_K, pask_K, pask_D, rasta);

      end;

  end; { max_srautas }
  {======= Maksimalaus srauto paieškos procedūrų pabaiga ===========}
{--------------- Pagrindinė programos dalis ------------------------}
begin
  įvesti (g, n, m);

  max_srautas (K, D);

  papildyti (K, D); { papildomas sprendinys }
  išvesti (D);

end.

